Programming Language Fundamentals (PLaF) Notes

v0.1089 - April 27, 2022

Eduardo Bonelli

Contents

Pr	Preface 7										
1	A Ca	A Calculator									
	1.1	Syntax									
		1.1.1	Concrete Syntax								
		1.1.2	Abstract Syntax								
	1.2	Interpre	eter	. 10							
		1.2.1	Specification	. 11							
		1.2.2	Implementation	. 13							
			1.2.2.1 Implementation: Final	. 14							
	1.3	Exercis	ses	. 17							
2	Simp	ole Fun	nctional Languages	19							
	2.1	LET .		. 19							
		2.1.1	Concrete Syntax	. 19							
		2.1.2	Abstract Syntax	. 19							
		2.1.3	Environments	. 20							
		2.1.4	Interpreter	. 20							
			2.1.4.1 Specification	. 20							
			2.1.4.2 Implementation: Attempt I	. 21							
			2.1.4.3 Weaving Environments	. 24							
			2.1.4.4 Implementation: Final								
		2.1.5	Inspecting the Environment	. 30							
	2.2	Exercis	Ses	. 31							
	2.3	PROC									
	2.0	2.3.1	Concrete Syntax								
		2.3.2	Abstract Syntax	. 35							
		2.3.3	Interpreter								
		2.5.5	2.3.3.1 Specification								
			2.3.3.2 Implementation	. 36							
		2.3.4	•								
	2.4	-	Dynamic Scoping								
	2.4 Exercises										
	2.5	REC .		. 40							
		2.5.1	Concrete Syntax								
		ウトウ	Abetract Suntay	// 1							

4 CONTENTS

		2.5.3	Interpreter	2
			2.5.3.1 Specification	
			2.5.3.2 Implementation	
			2.3.3.2 Implementation	_
3	Imne	erative	Programming 4	7
•	3.1		e Data Structures in OCaml	
	5.1	3.1.1		•
		3.1.2	· · · · · · · · · · · · · · · · · · ·	•
	2.0	-	······································	•
	3.2		CIT-REFS	
		3.2.1	Concrete Syntax	
		3.2.2	Abstract Syntax	
		3.2.3	Interpreter	9
			3.2.3.1 Specification	9
			3.2.3.2 Implementing Stores	1
			3.2.3.3 Implementation	2
		3.2.4	Extended Example: Encoding Objects	4
	3.3	IMPI I	CIT-REFS	
	0.0	3.3.1	Concrete Syntax	
		3.3.2	Abstract Syntax	
		3.3.3	•	
		3.3.3	•	
			3.3.3.1 Specification	
			3.3.3.2 Implementation	
		_	3.3.3.3 letrec Revisited	
	3.4		eter Passing Methods	
		3.4.1	Call-by-Value	
		3.4.2	Call-by-Reference	9
			3.4.2.1 Modifying the Interpreter	9
		3.4.3	Call-by-Name	0
		3.4.4	Call-by-Need	1
	3.5	Exercis	es	2
4	Туре	es	6	7
	4.1		KED	7
		4.1.1	Concrete Syntax	7
		4.1.2	Abstract Syntax	
		4.1.3	Type-Checker	
		4.1.5	4.1.3.1 Specification	
			4.1.3.2 Towards and Implementation	
			4.1.3.3 Implementation	
		4.1.4	Adding Letrec	
		4.1.5	Exercises	2
_				_
5	Mod		77	
	5.1	Syntax		
		5.1.1	Concrete Syntax	7
		5.1.2	Abstract Syntax	8
	5.2	Interpre	eter	8
		•		

CONTENTS	<u> </u>
CONTENTS	J

В	Solution to Selected Exercises							87									
Α		porting File St	Files ructure														85 85
		Type-C 5.3.1 5.3.2	Checking Specification Implements Reading .	on ation			 		 	 			 		 		81 81 82
			Specification Implementa														

6 CONTENTS

Preface

These course notes provide supporting material for CS496 and CS510. They draw from ideas in [Fri08] but have been evolving over the semesters.

Chapter 1

A Calculator

We introduce a toy language called ARITH. In ARITH programs are simple arithmetic expressions. The objective of this chapter is to provide a gentle introduction to various concepts we will be developing later in these notes. These concepts include the syntax of a language and how its programs are executed.

1.1 Syntax

This section presents the syntax of ARITH. The syntax determines what sequences of symbols which make up our code counts as a syntactically correct program. The syntax is typically presented in the form of a grammar and referred to as the concrete syntax. There are many details in the concrete syntax that are irrelevant for executing a program. For example, an expression such as 4 / 2 might denote a program that divides 4 by 2. But one may also use an expression such as 4 div 2 to denote the same program. Which of these two is considered syntactically correct is determined by the concrete syntax. In order to execute the program, all we need to know is that 4 is being divided by 2, regardless of how the language requires you to write the division operator itself. The abstract syntax of a language is the underlying representation of a syntactically correct program, once we abstract away any inessential, concrete details. It typically takes the form of a tree and is referred to as an Abstract Syntax Tree. A program called a parser, receives a sequence of symbols and determines whether it conforms to the rules of the concrete syntax. If it doesn't it reports a "syntax error"; if it does, it produces an abstract syntax tree. We next address these topics in further detail for ARITH.

1.1.1 Concrete Syntax

The grammar below specifies the concrete syntax of ARITH. It determines what expressions are syntactically correct ARITH programs. Each line is called a **production**. Expressions enclosed in angle brackets are called **non-terminals**. The grammar below only has two non-terminals, $\langle \text{Expression} \rangle$ and $\langle \text{Number} \rangle$. Among all non-terminals one singles out the so called **start non-terminal**. In our case, the start non-terminal is $\langle \text{Expression} \rangle$. Symbols that appear to the right of "::=" and that are not non-terminals are called **terminals**. The grammar below has the following set of terminals: $\{-,/,(,)\}$. Note that we have not specified what terminals are generated by

the $\langle Number \rangle$ non-terminal; we will assume these to be all integers. Negative numbers will be required to be written between parenthesis.

```
\langle \text{Expression} \rangle ::= \langle \text{Number} \rangle
\langle \text{Expression} \rangle ::= \langle \text{Expression} \rangle - \langle \text{Expression} \rangle
\langle \text{Expression} \rangle ::= \langle \text{Expression} \rangle / \langle \text{Expression} \rangle
\langle \text{Expression} \rangle ::= \langle \text{Expression} \rangle
```

For the sake of simplicity, our language ARITH only supports two arithmetic operations, subtraction and division. We will later add others.

Examples of syntactically correct expressions are 3-4, (4/0)-4), 3-4-1, and (-4)/2. Examples of syntactically incorrect expressions are 3-4, $4 ext{ div } 2$, and 3-().

1.1.2 Abstract Syntax

Given a string of terminals s, a parser will produce an abstract syntax tree (AST) for s if it is a syntactically correct ARITH expression, otherwise it will fail with an error message. The AST is a value of type expr:

This figure is an example of a <u>code listing</u>. We occasionally add an indication to code listings as to where the snippet of code <u>resides</u>. For example, in this case it resides in file <u>ast.ml</u>.

1.2 Interpreter

An interpreter¹ is a process that, given an expression, produces the result of its evaluation. The implementation of interpreters in these notes will be developed in two steps, first we specify the interpreter and then we implement it proper.

- Specification of the interpreter. This consists in first providing a precise description of the
 possible <u>results</u> which a program can evaluate to. Then introducing <u>evaluation judgements</u>
 that state what result a program evaluates to. Finally, a set of <u>evaluation rules</u> is introduced
 that define the meaning of evaluation judgements by describing the behavior of each
 construct in the language.
- Implementation of the interpreter. Using the evaluation rules of the specification as a guide, an implementation is presented. The time invested in producing the evaluation rules in the previous step, betters our understanding of the interpreter's behavior and hence diminishes the chances of introducing errors when it is implemented.

To illustrate this approach, we next specify an interpreter for ARITH and then implement it.

¹We will use the words "interpreter" and "evaluator" interchangeably.

1.2.1 Specification

We begin by stating the possible results that can arise out of the evaluation of programs in ARITH. For now, we fix the set of **results** to be the integers \mathbb{Z} since evaluation of ARITH programs produce integers:

$$\mathbb{R} := \mathbb{Z}$$

The ":=" symbol is used for definitional equality, meaning here that \mathbb{R} is "defined to be" the set \mathbb{Z} of integers. We continue with the specification of the interpreter for ARITH by introducing evaluation judgements. **Evaluation judgements** are expressions of the form:

$$e \downarrow \!\!\downarrow n$$

where e is an expression in ARITH in abstract syntax and n is a result (i.e. n is an integer). Evaluation judgements are read as, "expression e evaluates to the integer n". The meaning of e ψ n is established via so called **evaluation rules**. The preliminary set of evaluation rules of ARITH are as follows:

$$\frac{-\frac{1}{\mathrm{Int(n)} \Downarrow n} \, \mathsf{EInt}}{\frac{\mathsf{e1} \Downarrow m \quad \mathsf{e2} \Downarrow n \quad p = m - n}{\mathrm{Sub(e1,e2)} \Downarrow p} \, \mathsf{ESub}} \, \frac{\mathsf{e1} \Downarrow m \quad \mathsf{e2} \Downarrow n \quad p = m/n}{\mathrm{Div(e1,e2)} \Downarrow p} \, \mathsf{EDiv}$$

Judgements above the horizontal line in an evaluation rule are called **hypotheses** and the one below is called the **conclusion**. A rule that does not have hypotheses is called an **axiom rule** or just axiom. Hypotheses of an evaluation rule are read from left to right. In particular, evaluation of the arguments of all arithmetic operations proceeds from left to right. It could have been stated in the opposite order from right-to-left and, at this point in time, does not make much of a difference². A **derivation tree** is a tree of evaluation judgements whose nodes are instances of evaluation rules and, moroever, whose leaves are instances of axioms. A judgement $\mathbf{e} \Downarrow n$ is **derivable or is said to 'hold'** if there is a derivation tree with $\mathbf{e} \Downarrow n$ as its root.

Example 1.2.1. For example Sub(Int 3, Int 1) \Downarrow 2 is a derivable evaluation judgement:

$$\frac{\overline{\mathit{Int}\,(3) \Downarrow 3}}{\mathit{Sub}\,(\mathit{Int}\,\,3,\mathit{Int}\,\,1) \Downarrow 2} \, \mathsf{EInt} \qquad 2 = 3 - 1 \\ \mathit{Sub}\,(\mathit{Int}\,\,3,\mathit{Int}\,\,1) \Downarrow 2$$

The evaluation judgement $Sub(Int 3, Int 1) \downarrow 1$ is not derivable. The evaluation judgement $Sub(Div(Int 8, Int 2), Int 1) \downarrow 3$ is also derivable.

We are not quite done with the task of specifying our interpreter since not all expressions in ARITH return numbers. For example, Div(Int 2,Int 0) does not evaluate to any number. Rather it should evaluate to an error. Thus, the above mentioned evaluation rules are incomplete since there are situations that are left unspecified. It is important for have a complete set of rules so that when we implement the interpreter there are no ambiguities. Moreover, we must revisit our notion of result since it should include an error as a possible outcome. A result is either an integer or a special element error:

 $^{^2}$ But later in our development, when evaluation of expressions can cause certain effects (such as modifying mutable data structures), this difference will become relevant.

$$\frac{-1 \Downarrow m \quad e2 \Downarrow n \quad p = m - n}{\text{Sub}(\text{e1},\text{e2}) \Downarrow p} \text{ESub} \quad \frac{\text{e1} \Downarrow m \quad \text{e2} \Downarrow n \quad n \neq 0 \quad p = m/n}{\text{Div}(\text{e1},\text{e2}) \Downarrow p} \text{EDiv}$$

$$\frac{-1 \Downarrow error}{\text{Sub}(\text{e1},\text{e2}) \Downarrow error} \text{ESubErr1} \quad \frac{\text{e1} \Downarrow m \quad \text{e2} \Downarrow error}{\text{Sub}(\text{e1},\text{e2}) \Downarrow error} \text{ESubErr2}$$

$$\frac{-1 \Downarrow error}{\text{Sub}(\text{e1},\text{e2}) \Downarrow error} \text{EDivErr1} \quad \frac{-1 \Downarrow m \quad \text{e2} \Downarrow error}{\text{Div}(\text{e1},\text{e2}) \Downarrow error} \text{EDivErr2} \quad \frac{-1 \Downarrow m \quad \text{e2} \Downarrow 0}{\text{Div}(\text{e1},\text{e2}) \Downarrow error} \text{EDivErr3}$$

Figure 1.1: Evaluation rules for ARITH

$$\mathbb{R} := \mathbb{Z} \cup \{error\}$$

The subset of results that are integers are called <u>expressed values</u>: it is the name given to non-error results of evaluation. The previously introduced evaluation judgements are thus revisited below. The final form that evaluation judgements take for ARITH are:

$$\mathbf{e} \Downarrow r$$

where r denotes a result of the computation $r \in \mathbb{R}$.

The evaluation rules defining this new judgement, and therefore the evaluation rules for ARITH, are those presented in Figure 1.1, where $m,n,p\in\mathbb{Z}$. The first three rules are the ones already presented above. Rules ESubErr1, ESubErr2, EDivErr1, and EDivErr2 state how error propagation takes place. The last one introduces a new error, namely division by zero. Moving forward, and for the sake of brevity, we will not be specifying the error propagation rules when specifying the interpreter. We will only be presenting the error introduction rules.

Example 1.2.2. The evaluation judgement $Sub(Div(Int\ 8,Int\ 0),Int\ 1) \Downarrow error$ is derivable. Just like in the previous example, the evaluation judgement $Sub(Int\ 3,Int\ 1) \Downarrow 1$ is not derivable.

We next address the implementation of an interpreter for ARITH. We will do so in two attempts, first a preliminary attempt and then a final one. The preliminary attempt has various drawbacks that we will point out along the way but has the virtue of serving as a convenient stepping stone towards the final one.



A summary of some important concepts we have covered are listed below. Make sure you look them up above:

- Expressed Value Evaluation Judgement
- Evaluation Rules
- Derivation Tree
- Derivable Evaluation Judgements

1.2.2 **Implementation**

In order to use the evaluation rules as a guideline for our implementation we first need to model both components of evaluation judgements in OCaml, namely expression e and result r in $e \downarrow r$. The former is already expressed in abstract syntax, which we encoded as the algebraic data type expr in OCaml. So that item has already been addressed. As for the latter, since it denotes either an integer or an error, we will model it in OCaml using the following type³:

```
type 'a result = Ok of 'a | Error of string
                                                                           ds.ml
```

For example, the type int result may be read as a type that states that "the result of the evaluator is an integer or an error". Likewise, bool result may be read as a type that states that "the result of the evaluator is a boolean or an error". In summary, a result can either be a meaningful value of type 'a prefixed with the constructor Ok, or else an error with an argument of type string prefixed with an Error constructor. For example, Ok 3 has type int result.



In a type expression such as int result, we say int is a "type" and int result is a "type". But we refer to result as a **type constructor** since, given a type 'a, it constructs a type 'a result.

We can now proceed with an implementation of an interpreter for ARITH following the evaluation rules as close as possible. If we call our evaluator function eval_expr, its type can be expressed as follows, indicating that evaluation consumes an expression and returns either an integer or an error with a string description:

```
eval_expr : exp -> int result
```

The code is given in Figure 1.2. The eval_expr function is defined by recursion over the structure of expressions in abstract syntax (i.e. values of type expr). In the first clause, Int(n), it simply returns Ok n. Note that returning just n would be incorrect since out interpreter must

 $^{^3}$ OCaml has a built-in type type ('a,'b) result = 0k of 'a | Error of 'b. We could have used this type but it is slightly more general than necessary since our errors will always be accompanied by a string argument rather than different types of arguments.

```
let rec eval_expr : expr -> int result =
     fun e ->
     match e with
       Int(n) -> 0k n
       Sub(e1,e2) ->
       (match eval_expr e1 with
6
          Error s -> Error s
        Ok m -> (match eval_expr e2 with
8
                     Error s -> Error s
                    | Ok n -> Ok (m-n)))
10
     | Div(e1,e2) ->
       (match eval_expr e1 with
12
          Error s -> Error s
          Ok m -> (match eval_expr e2 with
14
                    | Error s -> Error s
                    \mid 0k n \rightarrow if n==0
16
                               then Error "Division by zero"
                               else Ok (m/n))
18
                                                                              interp.ml
```

Figure 1.2: Preliminary Interpeter for ARITH

return a value of type int result, not of type int. In the clause for Sub(e1,e2), the match keyword forces evaluation of e1 before e2, as indicated by the evaluation rules⁴. A similar comment applies to the other arithmetic operation. One notices that a substantial amount of code checks for errors and then propagates them. Notice too that, in the Div case, in addition to propagating errors resulting from evaluating its arguments e1 and e2, it generates a new one if the denominator is 0. This is in accordance with the evaluation rule EDivErr3. The only error modeled in ARITH is division by zero. An example of error propagation takes place in an ARITH expression such a Add(Div(Int 1,Int 0),e), where e can be any expression. Here the expression e is never evaluated since evaluation of Div(Int 1,Int 0) produces Error "Division by zero", hence e is ignored and Error "Division by zero" is immediately produced as the final result of evaluation of the entire expression.

1.2.2.1 Implementation: Final

Although certainly necessary, there is no interesting computational content in error propagation. It would be best to have it be handled behind the scenes, by appropriate error propagation helper functions. We next introduce three helper functions for this purpose:

- return,
- error, and
- (>>=) (pronounced "bind").

The code for these functions is given in Figure 1.3. The return function simply returns its argument inside an Ok constructor and may thus be seen as producing a non-error result. A similar comment applies to error: given a string it produces an error result by simply prefixing the

⁴OCaml evaluates arguments from right to left.

```
let return : 'a -> 'a result =
    fun v -> 0k v

4 let error : string -> 'a result =
    fun s -> Error s

6 let (>>=) : 'a result -> ('a -> 'b result) -> 'b result =
    fun c f ->
    match c with
10 | Error s -> Error s
    | 0k v -> f v
ds.ml
```

Figure 1.3: The Error Monad

string with the Error constructor. The infix operator (>>=) is called <u>bind</u> and is left associative⁵. Consider the expression c >>= f; its behavior may be described as follows:

- 1. evaluate the argument c to produce a result (*i.e.* a non-error value or an error value); if c returns an error, propagate it and conclude.
- 2. otherwise, if c returns 0k v, for some expressed value v, then pass v on to f by evaluating f v.

An alternative description of these helper functions is as follows. Let us dub expressions of type int result, structured programs (we could have taken the more general type 'a result as our notion of structured programs, but the latter will suffice for our explanation). Structured programs may be seen as programs that, apart from producing an integer as end product, can manipulate additional structure such as error handling, state, non-determinism, etc. In our particular case, a structured program handles errors as additional structure. Under this light, we can describe the helper functions as follows:

- return may be seen as a function that creates a (trivial) structured program that returns an integer (i.e. non-error) result.
- error may be seen as a function that creates a (trivial) structured program that returns an error result.
- (>>=) may be seen as a means of composing structured programs. In c >>= f, structured program c is composed with structured program f v, where v is the non-error result of c. If c produces an error, then evaluation of f is skipped.

Let us rewrite our interpreter for our simple expression language using these helper functions.

```
let rec eval_expr : expr -> int result =
fun e ->
match e with
| Int(n) -> return n
| Sub(e1,e2) ->
```

⁵The precedence and associativity of user-defined infix/prefix operators may be consulted here: https://caml.inria.fr/pub/docs/manual-caml-light/node4.9.html

```
6     eval_expr e1 >>= (fun n1 ->
        eval_expr e2 >>= (fun n2 ->
8     return (n1-n2)))
| Div(e1,e2) ->
10     eval_expr e1 >>= (fun n1 ->
        eval_expr e2 >>= (fun n2 ->
12     if n2==0
        then error "Division by zero"
14     else return (n1/n2)))
interp.ml
```

Consider the code for the Sub(e1,e2) case. Notice how if eval_expr e1 produces an error result, say Error "Division by zero" because e1 had a divison by zero, then (>>=) simply ignores its second argument, namely the expression (fun n1 -> eval_expr e2 >>= (fun n2 -> return (n1+n2))), and returns the error result Error "Division by zero" immediately as the final result of the evaluation, thus effectively propagating the error.

In fact, we can further simplify this code by dropping superfluous parenthesis. This leads to our final evaluator for ARITH expressions.

```
let rec eval_expr : expr -> int result =
     fun e ->
2
     match e with
     Int(n) -> return n
     Sub(e1,e2) ->
       eval_expr e1 >>= fun n1 ->
       eval_expr e2 >>= fun n2 ->
       return (n1-n2)
     | Div(e1,e2) ->
       eval_expr e1 >>= fun n1 ->
10
       eval_expr e2 >>= fun n2 ->
       if n2==0
12
       then error "Division by zero"
       else return (n1/n2)
14
```

Some additional observations on the behavior of the error handling operations:

The symbol \simeq above means that the left and right hand sides of these equations behave the same way.



The result type, together with the operations return, error and (>>=) is called an **Error Monad**. Monads are well-known in pure functional programming languages like Haskell, where they allow to handle side-effects behind the scenes without compromising equational reasoning (see the equations presented above). However, they are also important in non-pure functional languages, like OCaml, where they are a means to better structure one's code, as we have seen from our use of it here.

1.3 Exercises

Exercise 1.3.1. What is the difference between a result and an expressed value?

Exercise 1.3.2. Consider the extension of ARITH with a new expression that returns the absolute value of the value of its argument. The concrete syntax of ARITH is extended with the production:

```
\langle \mathsf{Expression} \rangle ::= abs(\langle \mathsf{Expression} \rangle)
```

and the abstract syntax with a new constructor:

- 1. Do the set of results in this extended language need to be modified? Think about whether new errors are introduced by the abs construct or whether new kinds of expressed values (other than integers) are possible.
- 2. Do evaluation judgements need to be modified? Think about whether evaluation judgements still have the form $e \downarrow r$, for e an expression in the extended language and r a result.
- 3. Add the two evaluation rules for the new language construct abs(e). You may assume that abs is the name of the mathematical function that returns the absolute value of an integer.
- 4. Extend the interpreter eval_expr to handle this case.

Chapter 2

Simple Functional Languages

2.1 LET

2.1.1 Concrete Syntax

```
 \begin{array}{lll} \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Number} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Identifier} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Expression} \rangle \langle \mathsf{BOp} \rangle \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{zero?}(\langle \mathsf{Expression} \rangle) \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{if} \langle \mathsf{Expression} \rangle \mathsf{then} \langle \mathsf{Expression} \rangle \mathsf{else} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{let} \langle \mathsf{Identifier} \rangle = \langle \mathsf{Expression} \rangle \mathsf{in} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{(Expression)} \rangle \\ \langle \mathsf{BOp} \rangle & ::= & \mathsf{+} | - | * | / \end{aligned}
```

Note that we have not specified what terminals are generated by the $\langle Identifier \rangle$ non-terminal; we will assume these to be sequences of symbols 'a'-'z', 'A'-'Z', '0'-'9', or '_' that start with a letter.

2.1.2 Abstract Syntax

2.1.3 Environments

Consider the LET expression x+2. Variables such as x are referred to as <u>identifiers</u>. We cannot determine the result of evaluating this expression because, in a sense, it is incomplete. Indeed, unless we are given the value assigned to the identifier x, we cannot determine whether evaluation of x+2 should result in an error (if say, x held the value true¹), or a number such as 4 (if, say, x held the value 2). Therefore, evaluation of expressions in LET require an assignment of values to identifiers. These assignments are called <u>environments</u>. The interpreters developed in these notes are therefore known as <u>environment-based interpreters</u> as opposed to <u>substitution-based interpreters</u>. In the latter values are substituted directly into the expressions rather than recording, and then looking them up, in environments.

An **environment** is a partial function from the set of identifiers to the set of expressed values. **Expressed values**, denoted \mathbb{EV} , are the set of values that are not errors that we can get from evaluating expressions. In ARITH the only expressed values are the integers. In LET they are the integers and the booleans:

$$\mathbb{EV} := \mathbb{Z} \cup \mathbb{B}$$

where $\mathbb{B} := \{true, false\}$. If \mathbb{ID} denotes the set of all identifiers², then we can define the set of all environments \mathbb{ENV} as follows.

$$\mathbb{ENV} := \mathbb{ID} \rightharpoonup \mathbb{EV}$$

We use letters ρ and ρ' to denote environments. For example, $\rho = \{x := 1, y := 2, z := true\}$ is an environment that assigns 1 to x, 2 to y and true to z. We write $\rho(id)$ for the value associated to the identifier id. For example, $\rho(x)$ is 1.

2.1.4 Interpreter

2.1.4.1 Specification

As you might recall from our presentation of ARITH, evaluation of an ARITH expression produces a <u>result</u>. A result could either be an integer or an error. We can still get an error from evaluating a LET expression since ARITH expressions are included in LET expressions. However, if there is no error, then in LET we can either get an integer or a boolean. The set of results for LET is thus:

$$\mathbb{R} := \mathbb{EV} \cup \{error\}$$

where \mathbb{EV} was updated above during our discussion on environments. Evaluation judgements for LET include an environment:

$$e, \rho \downarrow r$$

It should be read as follows, "evaluation of expression e under environment ρ , produces result r". The rules defining this judgement are presented in Figure 2.1. The last four rules handle error generation, the first eight handle standard (i.e. non-error) evaluation. The rules for error propagation are omitted. Rule Elnt is the same as in ARITH, except that the judgement has

¹We do not have booleans in ARITH but we will in LET.

²The elements of \mathbb{ID} are assumed to be the same as those generated by the non-terminal (Identifier).

$$\frac{\operatorname{el},\rho \Downarrow m \quad \operatorname{e2},\rho \Downarrow n \quad n \neq 0 \quad p = m/n}{\operatorname{Div}(\operatorname{el},\operatorname{e2}),\rho \Downarrow p} \operatorname{EDiv}$$

$$\frac{\operatorname{e},\rho \Downarrow v \quad v = 0}{\operatorname{IsZero}(\operatorname{e}),\rho \Downarrow true} \operatorname{EIZTrue} \quad \frac{\operatorname{e},\rho \Downarrow v \quad v \in \mathbb{Z} \quad v \neq 0}{\operatorname{IsZero}(\operatorname{e}),\rho \Downarrow true} \operatorname{EIZETrue} \quad \frac{\operatorname{el},\rho \Downarrow v \quad v \in \mathbb{Z} \quad v \neq 0}{\operatorname{IsZero}(\operatorname{el}),\rho \Downarrow false} \operatorname{EIZETrue} \quad \frac{\operatorname{el},\rho \Downarrow false \quad \operatorname{e3},\rho \Downarrow v}{\operatorname{ITE}(\operatorname{el},\operatorname{e2},\operatorname{e3}),\rho \Downarrow v} \operatorname{EITEFalse} \quad \frac{\operatorname{el},\rho \Downarrow false \quad \operatorname{e3},\rho \Downarrow v}{\operatorname{ITE}(\operatorname{el},\operatorname{e2},\operatorname{e3}),\rho \Downarrow v} \operatorname{ELEE} \quad \frac{\operatorname{el},\rho \Downarrow v \quad \operatorname{e2},\rho \oplus \{\operatorname{id}:=w\} \Downarrow v}{\operatorname{Let}(\operatorname{id},\operatorname{el},\operatorname{e2}),\rho \Downarrow v} \operatorname{ELEE} \quad \frac{\operatorname{el},\rho \Downarrow m \quad \operatorname{e2},\rho \Downarrow 0}{\operatorname{Div}(\operatorname{el},\operatorname{e2}),\rho \Downarrow error} \operatorname{EDivErr} \quad \frac{\operatorname{el},\rho \Downarrow v \quad v \notin \mathbb{B}}{\operatorname{ITE}(\operatorname{el},\operatorname{e2},\operatorname{e3}),\rho \Downarrow error} \operatorname{EIZErr} \quad \frac{\operatorname{el},\rho \Downarrow v \quad v \notin \mathbb{B}}{\operatorname{ITE}(\operatorname{el},\operatorname{e2},\operatorname{e3}),\rho \Downarrow error} \operatorname{EITEErr} \quad \operatorname{EITEERRR} \quad \operatorname{EITEERRR} \quad \operatorname{EITEERRR} \quad \operatorname{EITEERRR} \quad \operatorname{EITE$$

Figure 2.1: Evaluation Semantics for LET (error propagation rules omitted)

an environment (which plays no role in this rule). Rule EVar performs lookup in the current environment. The related rule EVarErr models the error resulting from lookup failing to find the identifier in the environment. The rules for addition, subtraction and multiplication are similar as the one for division and omitted. The notation $\rho \oplus \{ \mathrm{id} := w \}$ used in ELet stands for the environment that maps expressed value w to identifier id and behaves as ρ on all other identifiers.

2.1.4.2 Implementation: Attempt I

Before implementing the evaluator we must first implement expressed values and environments. Expressed values can be naturally described using algebraic data types. Environments can be modeled in various ways in OCaml: as functions, as association lists, as hash tables, as algebraic data types, etc. Due to its simplicity we follow the last of these.

Operations on environments are:

```
let rec eval_expr : expr -> env -> exp_val result =
    fun e en ->
     match e with
     Int(n) -> return (NumVal n)
     Var(id) -> apply_env id en
     | Div(e1,e2) \rightarrow (* Add, Sub and Mul are similar and omitted *)
6
       eval_expr e1 en >>=
       int_of_numVal >>= fun n1 ->
8
       eval_expr e2 en >>=
       int_of_numVal >>= fun n2 ->
10
      if n2==0
      then error "Division by zero"
12
       else return (NumVal (n1/n2))
     | IsZero(e) ->
14
       eval_expr e en >>=
       int_of_numVal >>= fun n ->
16
       return (BoolVal (n = 0))
     | ITE(e1,e2,e3) ->
18
       eval_expr e1 en >>=
       bool_of_boolVal >>= fun b ->
20
       if b
       then eval_expr e2 en
22
       else eval_expr e3 en
     Let(id,def,body) ->
       eval_expr def en >>= fun ev ->
       eval_expr body (extend_env en id ev)
     | _ -> failwith "Not implemented yet!"
                                                                            interp.ml
```

Figure 2.2: Preliminary Interpeter for LET

```
1 let empty_env : unit -> env =
    fun () -> EmptyEnv
3
2 let extend_env : env -> string -> exp_val -> env =
5    fun env id v -> ExtendEnv(id,v,env)

7 let rec apply_env : string -> env -> exp_val result =
    fun id env ->
    match env with
    | EmptyEnv -> error (id^" not found!")
11    | ExtendEnv(v,ev,tail) ->
        if id=v
        then return ev
        else apply_env id tail
ds.ml
```

Notice that apply_env en id has exp_val result as return type because it returns an error if id is not found in the environment en. However, if there is an expressed value associated to id in en, then that will be returned (wrapped with an Ok constructor).

Next we implement an interpreter for LET by following the evaluation rules of Figure 2.1 closely. Indeed, the evaluation rules shall serve as a specification for, and thus guide, our implementation. The code itself is given in Figure 2.2. The case for Var(id) simply invokes apply_env to look up the expressed value associated to the identifier id in the environment env.

The case for Div(e1,e2) makes use of an auxiliary operation int_of_numVal, explained below.

```
let int_of_numVal : exp_val -> int result =
fun ev ->
match ev with
| NumVal n -> return n
| _ -> error "Expected a number!"
ds.ml
```

Evaluation of e1 in Div(e1,e2) could produce an expressed value other than a number (i.e. other than a NumVal). The function int_of_numVal checks to see whether its argument is a NumVal or not, returning a result that consists of an error, if it is not, or else the number itself (without the NumVal tag). This number is then bound to the variable n1. A similar description determines the value of n2. Finally, if n2 is zero, an error is returned, otherwise the desired quotient is produced as a result.

The code for the other binary operators, for the zero predicate and for the conditional are similar. The case for ITE uses a similar helper function bool_of_boolVal. The last case, Let, evaluates the definition and then extends the environment appropriately before evaluating the body. Notice how local scoping is implemented by adding a new entry into the environment.



The default case, in the last line of Figure 2.2, handles language constructs that the parser supports but will only be explained later. This case simply reports that they are not implemented yet. Notice how the OCaml runtime Failure exception is used to report an error, rather than the error operation reserved for object-level (i.e. those arising from errors resulting from the execution of the LET) errors.

The top level function for the interpreter is called interp. It parses the string argument, evaluates producing a function c that awaits an environment, and then feeds that function the empty environment EmptyEnv.

```
let interp (s:string) : exp_val result =
let c = s |> parse |> eval_expr
in c EmptyEnv
interp.ml
```

Here is an example run of our interpreter³:

```
utop # interp "
2 let x=2
in let y=3
4 in x+y";;
- : exp_val Ds.result = Ok (NumVal 5)
6
utop # interp "
8 let x=2
in let y=0
10 in x+(x/y)";;
- : exp_val Ds.result = Error "Division by zero"
utop
```

 $^{^3}$ Negative numbers must be placed between parenthesis. For example, interp "(-7)" rather than interp "-7".



In ARITH there was only one possible error that could be generated and then propagated, namely the division by zero error. In LET there are four possible errors that can be generated and propagated: division by zero, identifier not found, expected a number and expected a boolean.

2.1.4.3 Weaving Environments

Our code for LET seems well-structured and robust enough to be extensible to additional language features. Even so, we can perhaps take a step further. Notice that the environment is explicitly threaded around the entire program. Indeed, consider the following excerpt from Figure 2.2 and notice how the environment (highlighted) is passed on to each occurrence of eval_expr.

```
let rec eval_expr : expr -> env -> exp_val result =

fun e en ->
match e with

i ...
ITE(e1,e2,e3) ->
eval_expr e1 en >>=
bool_of_boolVal >>= fun b ->

if b
then eval_expr e2 en
else eval_expr e3 en
```

The reason en is passed on in each case above, is that all expressions e1, e2 and e3 are evaluated under that <u>same</u> environment. This occurs with other language constructs too such as Add(e1,e2), Div(e1,e2), Sub(e1,e2) and Mul(e1,e2), where both e1 and e2 are evaluated under the environment en. An alternative would be to have the environment be passed around "behind the scenes", in the same way that error propagation is handled behind the scenes. The resulting code would look something like this, where all references to the environment have been removed, including the one on line 2:

```
let rec eval_expr : expr -> env -> exp_val result =

fun e en ->
match e with

i...
ITE(e1,e2,e3) ->
eval_expr e1 en >>=
bool_of_boolVal >>= fun b ->
if b
then eval_expr e2 en
else eval_expr e3 en

Listing 2.3: Naive removal of environment arguments
```

We would still need to provide an environment since eval_expr expects both expression and environment. That would be done by interp:

```
let interp (s:string) : exp_val result =
let c = s |> parse |> eval_expr
in c EmptyEnv

Listing 2.4: Naive removal of environment arguments
```

Unfortunately, the resulting code in Listing 2.3 doesn't type-check. Let us take a closer look at the bind operator used in line 6:

```
eval_expr e1 >>= ... (2.1)
```

Recall from Figure 1.3 that the type of (>>=) is

```
(>>=) : 'a result -> ('a -> 'b result) -> 'b result
```

The expression $eval_expr$ e1 in (2.1) is therefore expected to have type 'a result (where 'a can be any type, in particular exp_val). However, since we removed the environment argument it instead has type $env -> exp_val$ result. Indeed, $eval_expr$ e1 now produces a:

function that waits for the environment and then produces a result.

This means that bind now has to be able to compose "functions that wait for environments and produce a result" rather than composing "results". In other words, we have to put forward a new proposal for the type of bind:

```
Currently (>>=): 'a result -> ('a -> 'b result) -> 'b result

New proposal (>>=): (env -> 'a result) -> ('a -> (env -> 'b result)) -> (env -> 'b result).
```

Lets give the type env -> 'a result a name, so that we can improve legibility of the type expression above. Consider the following new ea_result type constructor, read "environment abstracted result", defined by simply abstracting the type of environments over the standard result type:

```
type 'a ea_result = env -> 'a result
```

Now we can apply this type synonym and recast our table above as:

```
Currently (>>=): 'a result -> ('a -> 'b result) -> 'b result

New proposal (>>=): 'a ea_result -> ('a -> 'b ea_result) -> 'b ea_result.
```

Of course, we'll need to update the code for bind (and the other helper functions). We will do sho shortly. Applying the type synonym again, this time to the type of eval_expr, the new type for our interpreter is now:

```
Currently eval\_expr : expr \rightarrow env \rightarrow exp\_val result New proposal eval\_expr : expr \rightarrow exp\_val ea\_result
```

Updating the helper functions. Since the type for the helper functions such as bind has changed, we must now update their code. The new code for them is in Figure 2.5. Function return v used to return v. But notice now how it returns a function that waits for an environment v and only then returns v. It may perhaps result odd that the environment seems not to be used for anything. However, other helper functions will make use of it (for example, (>>=)). Note also how (>>=) now passes the environment argument v first to v and then to v thus effectively threading the environment for us. You may safely ignore (>>+) for now, we'll explain it later. Also, we have a new operation run that given an environment abstracted result, will feed it the empty environment and thus perform the computation itself resulting in either an ok value or an error value. It is essentially the same as Listing 2.4 except that, since this function will be placed in the file interp.ml, it is best to avoid using the names of the constructors for environments.

```
type 'a result = Ok of 'a | Error of string
   type 'a ea_result = env -> 'a result
   let return : 'a -> 'a ea_result =
   fun v ->
    fun env -> Ok v
   let error : string -> 'a ea_result =
   fun s ->
    fun env -> Error s
   let (>>=) : 'a ea_result -> ('a -> 'b ea_result) -> 'b ea_result =
   fun c f ->
14
    fun env ->
    match c env with
16
    | Error err -> Error err
   | Ok v -> f v env
let (>>+) : env ea_result -> 'a ea_result -> 'a ea_result =
    fun c d ->
    fun env ->
    match c env with
    | Error err -> Error err
    Ok newenv -> d newenv
26
   let run : 'a ea_result -> 'a result =
   fun c -> c EmptyEnv
                                                                           ds.ml
```

Figure 2.5: The Reader and Error Monad Combined

```
let interp (e:string) : exp_val result =
let c = e |> parse |> eval_expr
in run c
interp.ml
```

The variable c is used as mnemonic for "computation" (also referred to as a "structured program") the program that results from evaluating the abstract syntax tree of e. The computation is executed by passing it the empty environment.

2.1.4.4 Implementation: Final

We next revisit our evaluator for LET, this time making use of our new environment abstracted result type. The code is given in Figure 2.6. We briefly comment on some of the variants.

The code for the Int(n) variant, remains unaltered:

```
Int(n) -> return (NumVal n)
```

Note, however, that return (NumVal n) now returns a function that given an environment, ignores it and simply returns Ok (NumVal n).

The Var(id) variant is similar, it is missing the environment:

```
| Var(id) -> apply_env id
```

Now apply_env is applied only to the argument id, thus producing an expression (through partial application) that waits for the second argument, namely the environment. This environment will be supplied when we run the computation (using run).

The variants Div(e1,e2), IsZero(e) and ITE(e1,e2,e3) are as in Figure 2.2 except that the environment argument has been dropped. Finally, consider Let(id,def,body). Let us recall from Figure 2.2, the code we had for this variant:

```
Let(id,def,body) ->
eval_expr def en >>= fun ev ->
eval_expr body (extend_env en id ev)
```

We first evaluate def in the current environment en producing an expressed value ev. This expressed value is used to extend the current environment ev, before evaluating the body body. Dropping the environment arguments, which are now threaded implicitly for us, results in:

```
Let(id,def,body) ->
eval_expr def en >>= fun ev ->
eval_expr body (extend_env id ev)
```

There are two problems with this code. First we need to be able to produce the modified environment resulting from adding the new key value-pair id:=ev into environment en **as a result** so that we can pass it on when evaluating body. This is achieved by updating extend_env, and empty_env too although we will not be needing the latter for now, that produces the updated environment as a result (notice the env in env ea_result):

```
let extend_env : string -> exp_val -> env ea_result =
fun id v ->
fun env -> Ok (ExtendEnv(id,v,env))
```

```
let rec eval_expr : expr -> exp_val ea_result =
    fun e ->
     match e with
     | Int(n) -> return (NumVal n)
     Var(id) -> apply_env id
    | Div(e1,e2) -> (* Add, Sub and Mul are similar and omitted *)
6
       eval_expr e1 >>=
      int_of_numVal >>= fun n1 ->
8
      eval_expr e2 >>=
      int_of_numVal >>= fun n2 ->
10
       if n2==0
       then error "Division by zero"
       else return (NumVal (n1/n2))
     | IsZero(e) ->
       eval_expr e >>=
       int_of_numVal >>= fun n ->
16
       return (BoolVal (n = 0))
    | ITE(e1,e2,e3) ->
18
      eval_expr e1 >>=
      bool_of_boolVal >>= fun b ->
20
       if b
      then eval_expr e2
22
       else eval_expr e3
    | Let(id,def,body) ->
       eval_expr def >>=
       extend_env id >>+
       eval_expr body
     | _ -> failwith "Not implemented yet!"
30 let parse s =
     let lexbuf = Lexing.from_string s in
     let ast = Parser.prog Lexer.read lexbuf in
32
     ast
   let interp (e:string) : exp_val result =
    let c = e |> parse |> eval_expr
    in run c
                                                                         interp.ml
```

Figure 2.6: Evaluator for LET

With this new operation we can produce the following code which is almost correct; we still have to discuss what to put in place of >>????:

```
Let(id,def,body) ->
2    eval_expr def >>= fun ev ->
    extend_env id ev >>???
4    eval_expr body
```

which can be simplified to

```
Let(id,def,body) ->
eval_expr def >>=
extend_env id >>???
eval_expr body
```

This code evaluates def under the current environment threaded by bind, then feeds the resulting expressed value into extend_env id to produce an extended environment. But now we are faced with a second problem. It is this extended environment that should be fed into eval_expr body and **not** the current environment that is threaded by bind (the current environment presumably has no mapping for id). This suggests introducing the following environment update operation:

```
let (>>+) : env ea_result -> 'a ea_result -> 'a ea_result =

fun c d ->
fun env ->

match c env with
| Error err -> Error err
| Ok newenv -> d newenv
```

An expression such as c >>+ d first evaluates c env, where env is the current environment, producing an environment newenv as a result, and then completely ignores the current environment env feeding that new environment as current environment for d.

With the help of environment update, we can now complete our code for Let:

```
Let(id,def,body) ->
2     eval_expr def >>=
     extend_env id >>+
4     eval_expr body
```



You can think of c1 (>>=) f as a form of composition of computations, "given an environment en, pass it on to c1 producing an expressed value v, then pass v and en on to f, and return its result as the overall result". While c1 (>>+) c2 may be thought of as, "given an environment en, pass it on to c1 producing an environment newenv (not an expressed value!) as a result which is passed on to computation c2, returning the latter's result as the result of the overall computation."

Note the absence of all references to $_{en}$ in Listing. 2.6. Indeed, the environment will be passed around when we execute $_{run}$ c. According to the definition of $_{run}$, $_{run}$ c just applies c to the empty environment $_{EmptyEnv}$.

Example 2.1.1. We conclude this section with some examples of expressions whose type involve the ea_result type constructor:

Expression return (NumVal 3)	Type exp_val ea_result	Informal Description Denotes a function that when given an environment, ignores it, and immediately returns Ok (NumVal 3).
error "oops"	'a ea_result	Denotes a function that when given an environment, ignores it, and immediately returns Error "oops".
apply_env "x"	exp_val ea_result	Denotes a function that when given an environment, inspects it to find the expressed value v associated to v . If it finds it, it returns v , otherwise it returns v and v and v and v .
extend_env <mark>"x"</mark> (NumVal 3)	env ea_result	Denotes a function that when given an environment, extends it producing a new environment newenv, with the new key-value pair x:=NumVal 3, and returns Ok newenv.

2.1.5 Inspecting the Environment

It is often convenient to be able to inspect the contents of the environment as a means of understanding how evaluation works or simply for debugging purposes. This section extends LET with a new expression debug(e) whose evaluation will print the contents of the current environment, ignoring e and halting evaluation with an error message. We first add a new production to the grammar defining the concrete syntax of LET:

```
\langle Expression \rangle ::= debug(\langle Expression \rangle)
```

We next add a new variant to the type expr defining the abstract syntax of LET:

```
type expr =
   ...
| Debug of expr
```

The next step is to specify, and then implement, the extension to the interpreter for LET that handles the new construct. What should we choose as the value resulting from evaluating Debug(e)? In other words, what should we choose to replace the questions marks below with?

$$\frac{\text{Debug(e)}, \rho \Downarrow ???}{\text{Debug}}$$
 EDebug

Since Debug(e) has to halt execution (and print the environment), we will have it return an error. This way, no matter where it is placed, the error will get propagated hence effectively halting all further execution. The evaluation rule EDebug becomes:

$$\frac{}{\texttt{Debug(e)}, \rho \Downarrow \textit{error}} \texttt{EDebug}$$

Finally, the implementation of this evaluation rule is given below. It makes use of an auxiliary function string_of_env, defined in ds.ml, which traverses an environment and returns a string representation of it.

```
eval_expr : expr -> exp_val ea_result =
fun e ->
match e with
...
| Debug(e) ->
string_of_env >>= fun str ->
print_endline str;
error "Debug called"
```

Note that there is a slight discrepancy between the specification of the evaluation rule describing how Debug(e) is evaluated (i.e. EDebug) and our implementation. Indeed, the latter prints two strings on the screen but the former does not mention any side-effects such as printing. The reason for this mismatch is that we have decided to keep the specification of our interpreters as simple as possible. In particular, we have decided not to model side-effects such as printing on the screen. Later we will show how to model other side-effecting operations when specifying interpreters. Notably, we will include an assignment operation in our language.

2.2 Exercises

Exercise 2.2.1 (\lozenge) . Write an OCaml expression of each of the types below:

```
1. expr
```

- 2. env
- $3. exp_val$
- 4. exp_val result
- 5. int result
- 6. env result
- 7. int ea_result
- 8. exp_val ea_result
- 9. env ea_result

Exercise 2.2.2. Consider the following code:

```
let c =
  empty_env () >>+
  extend_env "x" (NumVal 1) >>+
  extend_env "y" (BoolVal false) >>+
  string_of_env
```

where the helper function string_of_env is defined as follows:

- 1. Knowing that (>>+) associates to the left, fill in all the implicit parenthesis in the definition of c.
- 2. What is the type of c?
- 3. What happens when you load the code into utop and type c?
- 4. What happens when you load it into utop and type run c?

Exercise 2.2.3. Consider the following extension of LET with pairs. Its concrete syntax includes all the grammar productions of LET plus:

```
 \langle \mathsf{Expression} \rangle \ ::= \ pair(\langle \mathsf{Expression} \rangle, \langle \mathsf{Expression} \rangle) \\ | \ fst(\langle \mathsf{Expression} \rangle) \\ | \ snd(\langle \mathsf{Expression} \rangle)
```

Examples of programs in this language are

pair (2,3)
 pair (pair(7,9),3)
 pair(zero?(4),11-x)

4. snd(pair (pair(7,9),3))

The abstract syntax includes the following additional variants:

```
type expr =
...
| Pair of expr*expr
| Fst of expr
| Snd of expr
```

- 1. Specify the interpreter (i.e. its set of results and the new evaluation rules). You may assume that you have a product operation \times that computes the product of two sets.
- 2. Extend the implementation of eval_expr to handle the new language constructs. Are there any new errors?

Exercise 2.2.4. Consider another extension to LET with pairs. Its concrete syntax is:

```
 \langle \mathsf{Expression} \rangle \ ::= \ pair(\langle \mathsf{Expression} \rangle, \langle \mathsf{Expression} \rangle) \\ | \ unpair(\langle \mathsf{Identifier} \rangle, \langle \mathsf{Identifier} \rangle) = \langle \mathsf{Expression} \rangle \ in \langle \mathsf{Expression} \rangle
```

Pairs are constructed in the same way as in Exercise 2.2.3. However, to eliminate pairs instead of fst and snd we now have unpair. The expression unpair (x,y)=e1 in e2 evaluates e1, makes sure it is a pair with components v1 and v2 and then evaluates e2 in the extended environment where x is bound to v1 and v2. Examples of programs in this extension are the first three examples in Exercise 2.2.3 and:

- 1. unpair(x,y) = pair(3, pair(5, 12)) in x is a program that evaluates to 0k (NumVal 3).
- 2. The program let x = 34 in unpair (y,z)=pair(2,x) in z evaluates to 0k (NumVal 34).

The abstract syntax of this extension is:

```
type expr =
...
/ Pair of expr*expr
/ Unpair of string*string*expr*expr

ast.ml
```

Specify the interpreter (i.e. its evaluation rules) and then implement it.

Exercise 2.2.5. Consider the extension of LET with tuples. Its concrete syntax is that of LET together with the following new productions:

```
\langle \text{Expression} \rangle ::= \langle \langle \text{Expression} \rangle^{*(,)} \rangle
\langle \text{Expression} \rangle ::= untuple \langle | \text{Identifier} \rangle^{*(,)} \rangle = \langle \text{Expression} \rangle in \langle \text{Expression} \rangle
```

The *(,) above the nonterminal indicates zero or more copies separated by commas. The angle brackets construct a tuple with the values of its arguments. An expression of the form untuple < x1, ..., xn > = e1 in e2 first evaluates e1, makes sure it is a tuple of n values, say v1 to vn, and then evaluates e2 in the extended environment where each identifier xi is bound to vi. Examples of programs in this extension are:

- 1. <2,3,4>
- 2. <2,3,zero?(0)>
- 3. <<7,9>,3>
- 4. <zero?(4),11-x>
- 5. $untuple \langle x, y, z \rangle = \langle 3, \langle 5, 12 \rangle, 4 \rangle$ in x evaluates to 0k (NumVal 3).
- 6. let x = 34 in untuple $\langle y, z \rangle = \langle 2, x \rangle$ in z evaluates to 0k (NumVal 34).

Specify the interpreter (i.e. its evaluation rules) and then implement it.

Exercise 2.2.6. Consider the following extension of LET with records. Its concrete syntax is given adding the following new productions to that of LET:

```
 \begin{array}{lll} \langle \mathsf{Expression} \rangle & ::= & \{ \; \langle \mathsf{Identifier} \rangle = \langle \mathsf{Expression} \rangle^{+(;)} \} \\ \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Expression} \rangle. \langle \mathsf{Identifier} \rangle \\ \end{array}
```

Examples of programs in this extension are:

- 1. $\{age=2; height=3\}$
- 2. let $person = \{age=2; height=3\}$ in let $student = \{pers=person; cwid=10\}$ in student
- 3. $\{age=2; height=3\}$. age
- 4. $\{age=2; height=3\}$. ages
- 5. {aqe=2; aqe=3}

Assume that the expressed values of LET are extended so that now records of expressed values may be produced as a result of evaluating a program (see the examples above).

```
type exp_val =
...
/ RecordVal of (string*exp_val) list
```

The expr type encoding the AST is also extended:

```
type expr =
...
| Record of (string*expr) list
| Proj of expr*string
```

Specify the interpreter (i.e. its evaluation rules) and then implement it. Some examples of the result of evaluation of this extension are:

```
    {age=2; height=3}
    Should evaluate to Ok (RecordVal [("age", NumVal 2); ("height", NumVal 3)]).
```

2. let $person = \{age=2; height=3\}$ in let $student = \{pers=person; cwid=10\}$ in student

```
Should evaluate to Ok (RecordVal [("pers", RecordVal [("age", NumVal 2); ("height", NumVal 3)]); ("cwid", NumVal
```

- 3. $\{age=2; height=3\}.age$
 - Should evaluate to Ok (NumVal 2).
- {age=2; height=3}.ages
 Should evaluate to Error "Field not found".
- 5. {age=2; age=3}
 Should evaluate to Error "Record has duplicate fields".

2.3 PROC

This section adds first-class functions to LET.

2.3.1 Concrete Syntax

The concrete syntax for PROC consists in adding two new productions to the grammar of the concrete syntax for LET:

```
(Expression)
                  ::=
                         (Number)
                         (Identifier)
(Expression)
                  ::=
(Expression)
                  ::=
                          \langle Expression \rangle \langle BOp \rangle \langle Expression \rangle
(Expression)
                  ::= zero?((Expression))
(Expression)
                  ::= if (Expression) then (Expression) else (Expression)
(Expression)
                  ::= let (Identifier) = (Expression) in (Expression)
(Expression)
                  ::= (\langle \mathsf{Expression} \rangle)
(Expression)
                  ::= proc(\langle Identifier \rangle) \{\langle Expression \rangle\}
(Expression)
                  ::= (\langle Expression \rangle \langle Expression \rangle)
\langle BOp \rangle
                  ::= + | - | * | /
```

2.3.2 Abstract Syntax

```
type expr =
2   | Var of string
   | Int of int
4   | Add of expr*expr
   | Sub of expr*expr
6   | Mul of expr*expr
   | Div of expr*expr
8   | Let of string*expr*expr
   | IsZero of expr
10   | ITE of expr*expr
   | Proc of string*expr
12   | App of expr*expr
```

2.3.3 Interpreter

2.3.3.1 Specification

Evaluation judgements for PROC are exactly the same as for LET except that now the value resulting from evaluation of non-error computations, namely the expressed values, may either be an integer, a boolean or a **closure**. A closure is a triple consisting of an identifier, an expression and an environment. All three sets, expressed values, closures and environments must be defined mutually recursively since they depend on each other:

```
\begin{array}{lll} \mathbb{EV} & := & \mathbb{Z} \cup \mathbb{B} \cup \mathbb{CL} \\ \mathbb{CL} & := & \{ (\mathrm{id}, \mathrm{e}, \rho) \, | \, e \in \mathbb{EXP}, id \in \mathbb{ID}, \rho \in \mathbb{ENV} \} \\ \mathbb{ENV} & := & \mathbb{ID} \rightharpoonup \mathbb{EV} \end{array}
```

The evaluation judgement for PROC reads:

```
\mathbf{e},\rho \Downarrow r
```

The evaluation rules include those of LET (see Figure 2.1) plus the additional rules given in Figure 2.7.

$$\label{eq:proc_id_e} \begin{split} & \frac{}{\mathsf{Proc}(\mathtt{id},\mathtt{e}),\rho \Downarrow (\mathtt{id},\mathtt{e},\rho)} \mathsf{EProc} \\ & \frac{\mathtt{e1},\rho \Downarrow (\mathtt{id},\mathtt{e},\sigma) \quad \mathtt{e2},\rho \Downarrow w \quad \mathtt{e},\sigma \oplus \{\mathtt{id} := w\} \Downarrow v}{\mathsf{App}(\mathtt{e1},\mathtt{e2}),\rho \Downarrow v} \, \mathsf{EApp} \\ & \frac{}{\mathsf{app}(\mathtt{e1},\mathtt{e2}),\rho \Downarrow \mathit{error}} \, \mathsf{EAppErr} \\ \end{split}$$

Figure 2.7: Additional Evaluation rules for PROC (error propagation rules omitted)

2.3.3.2 Implementation

To extend the interpreter for LET to PROC, we need to model closures and then extend <code>eval_expr</code>. Modeling closures as runtime values is easy since closures are simply triples consisting of an identifier, an expression and an environment:

```
type exp_val =
    | NumVal of int
    | BoolVal of bool

    | ProcVal of string*Ast.expr*env
and
env =
    | EmptyEnv
| ExtendEnv of string*exp_val*env

ds.ml
```

Now, for eval_expr, we add code for two new variants in the definition of eval_expr, namely Proc(id,e) and App(e1,e2). Let us first analyze the former. Our first attempt might look something like this:

```
let rec eval_expr : expr -> exp_val ea_result =
fun e ->
match e with
| Proc(id,e) ->
return (ProcVal(id,e, en))
```

Evaluation of Proc(id,e) should produce a closure that includes both of id and e. It must also include the current environment, denoted en above. However, the identifier en is not in scope. Indeed, the current environment is passed around in the background by the helper functions for ea_result. We introduce a new helper function that reads the current environment and returns it as a result (i.e. Ok env, where env is the environment).

With this new function we can now implement the evaluator for Proc(id,e):

```
let rec eval_expr : expr -> exp_val ea_result =
fun e ->
```

```
match e with

| Proc(id,e) ->
    lookup_env >>= fun en ->

return (ProcVal(id,e,en))

interp.ml
```

Two alternative implementations for the Proc(id,e) case might be:

```
| Proc(id,e) ->
fun env -> return (ProcVal(id,e,env)) env
```

and

```
Proc(id,e) ->
fun env -> Ok (ProcVal(id,e,env))
```

This last one is perhaps the least recommendable since the constructor 0k should best not be used outside ds.ml.

We next consider the case for App(e1,e2). Evaluation of an application requires that we first evaluate e1 and make sure it is a closure. If that is the case, then clos will be bound to a triple containing its three components (parameter, body and environment). We then evaluate e2 and feed it to the helper function apply_clos clos (explained below), which does the rest of the job:

```
let rec eval_expr : expr -> exp_val ea_result =
fun e ->
match e with

| App(e1,e2) ->
eval_expr e1 >>=
clos_of_procVal >>= fun clos ->
eval_expr e2 >>=
apply_clos clos
```

The function clos_of_procVal is similar to int_of_numVal from Listing 2.2. It's code is:

```
let clos_of_procVal : exp_val -> (string*Ast.expr*env) ea_result =
fun ev ->
match ev with
| ProcVal(id,body,en) -> return (id,body,en)
| _ -> error "Expected a closure!"
ds.ml
```

The function $apply_clos$ sets en to be the new current environment and then extends it with the assignment of id to ev. Under this extended environment, it proceeds with the evaluation of the body of the closure, namely e.

```
let rec apply_clos : string*Ast.expr*env -> exp_val -> exp_val ea_result =
fun (id,e,en) ev ->
return en >>+
extend_env id ev >>+
eval_expr e
```

In passing we mention that the tuple type constructor has higher precedence than the function type constructor. Consequently, there is no need to place the type expression string*Ast.expr*env between parenthesis.

Listing 2.8 summarizes the code described above for procedures and applications.

```
let rec apply_clos : string*Ast.expr*env -> exp_val ->
     exp_val ea_result =
     fun (id,e,en) ev ->
     return en >>+
     extend_env id ev >>+
     eval_expr e
   and
     eval_expr : expr -> exp_val ea_result =
8
     match e with
10
     Proc(id,e)
       lookup_env >>= fun en ->
12
       return (ProcVal(id,e,en))
     | App(e1,e2) ->
14
       eval_expr e1 >>=
       clos_of_procVal >>= fun clos ->
16
       eval_expr e2 >>=
       apply_clos clos
18
                                                                           interp.ml
```

Figure 2.8: Interpreter for PROC

2.3.4 Dynamic Scoping

If we remove the line below, then we implement dynamic scoping:

```
let rec apply_clos : string*Ast.expr*env -> exp_val -> exp_val ea_result =
fun (id,e,en) ev ->
return en >>+
(extend_env id ev >>+
eval_expr e)
```

Indeed, in this case the environment that is extended by <code>extend_env</code> id a is the current environment and not the one that was saved in the closure. Here are some examples of executing programs in this variant of PROC:

```
interp "
2 let f = proc (x) { if zero?(x) then 1 else x*(f (x-1)) }
in (f 6) ";;
4 -: Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 720)
6 utop # interp "
let f = proc (x) { x+a }
8 in let a=2
in (f 2)";;
10 -: Ds.exp_val Ds.result = Ds.Ok (Ds.NumVal 4)
12 utop # interp "
let f = let a=2 in proc(x) { x+a}
14 in (f 2) ";;
-: Ds.exp_val Ds.result = Ds.Error "a not found!"
utop
```

2.4 Exercises

Exercise 2.4.1. Write a grammar derivation to show that let $f = proc(x) \{ x-11 \}$ in (f 77) is a valid program in PROC.

Exercise 2.4.2. Write down the parse tree for the expression let pred = proc(x) { x-1 } in (pred 5).

Exercise 2.4.3. Write down the result of evaluating the expressions below. Depict the full details of the closure, including the environment. Use the tabular notation seen in class to depict the environment.

- proc (x) { x-11 }
- proc (x) { let y=2 in x }
- let a=1 in proc (x) { x }
- let a=1 in let b=2 in proc (x) { x }
- let $f=(let b=2 in proc (x) {x}) in f$
- proc (x) { proc (y) { x-y } }

Exercise 2.4.4. Depict the environment extant at the breakpoint (signalled with the debug expression):

```
let a=1
in let b=2
in let c=proc (x) { x }
in debug((c b))
```

Exercise 2.4.5. Depict the environment extant at the breakpoint:

```
let a=1
in let b=2
in let c = proc (x) { debug(proc (y) { x-y } )}
in (c b)
```

Exercise 2.4.6. Depict the environment extant at the breakpoint:

```
let x=2
in let y=proc (d) { x }
in let z=proc(d) { x }
in debug(3)
```

Exercise 2.4.7. The result of evaluating the following expression is Ok (NumVal 4). Verify this.

```
let a = 3
in let p = proc (z) { z+a }
in let f = proc (x) { (p 1) }
in let a = 6
in (f 2)
```

Modify the interpreter so that it uses dynamic scoping rather than static scoping. Then evaluate the above expression again. What value does it return?

Exercise 2.4.8 (\Diamond). Use the "higher-order" trick of self-application to implement the mutually recursive definitions of even and odd in PROC:

```
even(0) = true
even(n) = odd(n-1)
odd(0) = false
odd(n) = even(n-1)
```

Exercise 2.4.9 (\Diamond). Use the "higher-order" trick of self-application to implement a function pbt that given a value v and a height h builds a perfect binary tree constructed out of pairs and that has v im the leaves and has height v. For example ($(pbt\ 2)\ 3)$ should produce

```
PairVal

(PairVal

(PairVal (PairVal (NumVal 2, NumVal 2),
PairVal (NumVal 2, NumVal 2)),
PairVal (PairVal (NumVal 2, NumVal 2),
PairVal (PairVal (NumVal 2, NumVal 2)),
PairVal

(PairVal (PairVal (NumVal 2, NumVal 2),
PairVal (NumVal 2, NumVal 2),
PairVal (NumVal 2, NumVal 2)),
PairVal (PairVal (NumVal 2, NumVal 2),
PairVal (PairVal (NumVal 2, NumVal 2),
PairVal (NumVal 2, NumVal 2))))
```

Lists and Trees

Exercise 2.4.10.

2.5 **REC**

Our language unfortunately does not support recursion⁴. The following attempt at defining factorial and then applying it to compute factorial of 5 fails. The reason is that f is not visible in the body of the proc.

```
let f =
    proc (x) {
        if zero?(x)
4        then 1
        else x*(f (x-1)) }
6 in (f 5)
```

In order to verify this, evaluate the following expression:

Note that the environment in the closure for f does not include a reference to f itself. The next language we shall look at, namely REC, includes a new programming abstraction that allows us to define recursive functions. In REC we will write:

⁴See exercises 2.4.8 and 2.4.9 on the "higher-order" trick though.

```
letrec fact(x) =
    if zero?(x)
    then 1
    else x * (fact (x-1))
in (fact 5)
```

2.5.1 Concrete Syntax

```
 \begin{array}{lll} \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Number} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Identifier} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \langle \mathsf{Expression} \rangle \langle \mathsf{BOp} \rangle \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{zero?}(\langle \mathsf{Expression} \rangle) \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{if} \langle \mathsf{Expression} \rangle \mathsf{then} \langle \mathsf{Expression} \rangle \mathsf{else} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{let} \langle \mathsf{Identifier} \rangle = \langle \mathsf{Expression} \rangle \mathsf{in} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{(Expression)} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{proc}(\langle \mathsf{Identifier} \rangle) \{ \langle \mathsf{Expression} \rangle \} \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{(Expression)} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{Expression} \rangle & ::= & \mathsf{letrec} \langle \mathsf{Identifier} \rangle (\langle \mathsf{Identifier} \rangle) = \langle \mathsf{Expression} \rangle \mathsf{in} \langle \mathsf{Expression} \rangle \\ \langle \mathsf{BOp} \rangle & ::= & \mathsf{+} | - | \mathsf{*} | / \\ \end{array}
```

2.5.2 Abstract Syntax

For example, the result of parsing the expression:

```
letrec fact(x) =
    if zero?(x)
then 1
    else x * (fact (x-1))
in (fact 5)

is the AST:

Letrec ("fact", "x",
    ITE (IsZero (Var "x"), Int 1,
    Mul (Var "x", App (Var "fact", Sub (Var "x", Int 1)))),
    App (Var "fact", Int 5))
```

$$\frac{\texttt{e2}, \rho \oplus \{\texttt{id} := (\texttt{par}, \texttt{e1}, \rho)^r\} \Downarrow v}{\texttt{Letrec(id,par,e1,e2)}, \rho \Downarrow v} \, \texttt{ELetRec}$$

$$\frac{\rho(\texttt{id}) = (\texttt{par}, \texttt{e}, \sigma)^r}{\texttt{Var(id)}, \rho \Downarrow (\texttt{par}, \texttt{e}, \sigma \oplus \{\texttt{id} := (\texttt{par}, \texttt{e}, \sigma)^r\})} \, \texttt{EVarLetRec}$$

Figure 2.9: Additional evaluation rules for REC

2.5.3 Interpreter

Recursive functions will be represented as special closures called recursion closures. Later we will look at another implementation involving circular environments. A **recursion closure** is a closure with a tag "r" to distinguish it from a standard closure, written $(id, e, \rho)^r$, where $e \in \mathbb{EXP}$, $id \in \mathbb{ID}$ and $\rho \in \mathbb{ENV}$. The set of all recursion closures is denoted \mathbb{RCL} :

```
\begin{array}{lll} \mathbb{E}\mathbb{N}\mathbb{V} &:= & \mathbb{ID} \rightharpoonup (\mathbb{E}\mathbb{V} \cup \mathbb{RCL}) \\ \mathbb{E}\mathbb{V} &:= & \mathbb{Z} \cup \mathbb{B} \cup \mathbb{CL} \\ \mathbb{CL} &:= & \{(\mathrm{id}, \mathrm{e}, \rho) \, | \, e \in \mathbb{E}\mathbb{XP}, id \in \mathbb{ID}, \rho \in \mathbb{E}\mathbb{NV}\} \\ \mathbb{R}\mathbb{CL} &:= & \{(\mathrm{id}, \mathrm{e}, \rho)^r \, | \, e \in \mathbb{E}\mathbb{XP}, id \in \mathbb{ID}, \rho \in \mathbb{E}\mathbb{NV}\} \end{array}
```

Note that recursion closures are not expressed values. We cannot write a program that, when evaluated, returns a recursion closure. They are an auxiliary device for defining evaluation of recursive programs. More precisely, recursive function definitions will be stored as recursion closures. However, lookup of recursive functions will produce standard closures, the latter being computed on the fly.

2.5.3.1 Specification

The set of results is the same as in PROC:

$$\mathbb{R} := \mathbb{EV} \cup \{error\}$$

Evaluation judgements for REC are the same as for PROC:

$$e, \rho \downarrow r$$

where $r \in \mathbb{R}$. Evaluation rules for REC are those of PROC together with the ones in Figure 2.9. Two new evaluation rules are added to those of PROC to obtain REC. The rule ELetRec creates a recursion closure and adds it to the current environment ρ and then continues with the evaluation of e2. The rule EVarLetRec does lookup of identifiers that refer to previously declared recursive functions. Upon finding the corresponding recursion closure in the current environment, it creates a new closure and returns it. Note that the newly created closure includes an environment that has a reference to f itself.

2.5.3.2 Implementation

Recursion closures are implemented by adding a new constructor to the type expr, namely ExtendEnvRec below:

Note that the arguments of <code>ExtendEnvRec(id,par,body,env)</code> are four: the name of the recursive function being defined <code>id</code>, the name of the formal parameter <code>par</code>, the body of the recursive function <code>body</code>, and the rest of the environment <code>env</code>. If we consider the environment $\rho \oplus \{ id := (par, e1, \rho)^r \}$ in the rule <code>ELetRec</code> of Figure 2.9, it would seem we are missing an argument. Indeed, the operator " \oplus " in the evaluation rule is modeled by the <code>ExtendEnvRec</code> constructor in our implementation. However, there is no need to store ρ in our implementation since it is just the tail of the environment.

Next we need an operation similar to <code>extend_env</code> but that adds a new recursion closure to the environment:

```
let extend_env_rec : string -> string -> Ast.expr -> env ea_result =
fun id par body ->
fun env -> Ok (ExtendEnvRec(id,par,body,env))

ds.ml
```

In addition, we need to update the implementation of apply_env so that it deals with lookup of recursive functions, thus correctly implementing EVarLetRec. This involves adding a new clause (see code highlighted below):

```
let rec apply_env : string -> exp_val ea_result =
     fun id -
     fun env ->
     match env with
     | EmptyEnv -> Error (id^" not found!")
     ExtendEnv(v,ev,tail) ->
       if id=v
       then Ok ev
8
       else apply_env id tail
10
       ExtendEnvRec(v,par,body,tail) ->
       if id=v
       then Ok (ProcVal (par,body,env))
12
       else apply_env id tail
                                                                                  ds.ml
```

Regarding the code for the interpreter itself, we need only add a new clause, namely the one for Letrec(id,par,e1,e2):

```
| Letrec(id,par,e1,e2) ->
extend_env_rec id par e1 >>+
eval_expr e2
interp.ml
```

Exercise 2.5.1. Evaluate the following expressions in utop:

```
1.
  utop # interp "
  let one=1
 in letrec fact(x) =
               if zero?(x)
                then one
               else x * (fact (x-1))
  in debug((fact 6))";;
2.
  utop # interp "
  let one=1
  in letrec fact(x) =
         debug(if zero?(x)
               then one
               else x * (fact (x-1))
  in (fact 6)";;
  utop # interp "
  let one=1
  in letrec fact(x) =
               if zero?(x)
                then one
                else x * (fact (x-1))
  in fact";;
```

Exercise 2.5.2 (\Diamond). Consider the following functions in OCaml:

```
let rec add n m =
     match n with
     / 0 -> m
     | n' -> 1 + add (n'-1) m
   let\ rec\ append\ l1\ l2\ =
     match l1 with
     / [] -> 12
     / h::t -> h :: append t 12
   let rec map l f =
11
     match l with
     / [] -> []
13
     / h::t -> (f h) :: map t f
15
   let rec filter l p =
17
     match l with
     / [] -> []
     / h::t ->
19
       if p h
        then h :: filter t p
21
        else filter t p
23
   let\ rec\ foldr\ l\ f\ a =
    match l with
25
      / [] -> a
     / h:: t \rightarrow f h (foldr t f a)
```

Code them in the extension of REC of Exercise 2.4.10. For example, here is the code for add:

Exercise 2.5.3 (\Diamond). Consider the following expression in REC

```
let z = 0

in let prod = proc (x) { proc (y) { x*y }}

in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))

in (f 10)
```

A debug instruction was placed somewhere in the code and it produced the environments below. Where was it placed? Identify and signal (see instructions below) the location for each of the three items below. Note that there may be more than one solution for each item, it suffices to supply just one.

Draw a box around the argument of debug:

```
let z=0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n)=if zero?(n) then 1 else ((prod\ n)\ (f\ (n-1)))
in (f\ 10)
```

```
2. 
>>Environment:
z:=NumVal 0
```

Draw a box around the argument of debug:

```
let z=0
in let prod=proc\ (x)\ \{\ proc\ (y)\ \{\ x*y\ \}\}
in letrec f(n)=if\ zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

Draw a box around the argument of debug:

```
let z=0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n)=if zero?(n) then 1 else ((prod\ n)\ (f\ (n-1)))
in (f\ 10)
```

Chapter 3

Imperative Programming

3.1 Mutable Data Structures in OCaml

This section discusses some OCaml language features that all data to be updated in-place. We will introduce references, arrays and mutable record fields.

3.1.1 A counter object

```
type counter =
    { inc: unit -> unit;
      dec: unit -> unit;
      read : unit -> int}
    let state = ref 0 in
     { inc = (fun () -> state := !state+1);
       dec = (fun () -> state := !state-1);
      read = (fun () -> !state) }
utop # c.read ();;
 - : int = 0
utop # c.inc ();;
 - : unit = ()
utop # c.read ();;
  - : int = 1
utop # c.dec ();;
 - : unit = ()
utop # c.read ();;
 -: int = 0
                                                                            utop
```

3.1.2 A stack object

```
type stack =
    { push : int -> unit;
     pop : unit -> int;
      top : unit -> int};;
 let s = let state = ref []
        in { push = (fun i -> state:= i::!state);
             pop = (fun () -> let temp = List.hd !state
                               in state := List.tl !state; temp);
              top = (fun () -> List.hd !state)}
utop # s.push 1;;
 - : unit = ()
utop # s.push 2;;
 - : unit = ()
utop # s.pop ();;
 - : int = 2
utop # s.top ();;
 -: int = 1
                                                                            utop
```

3.2 EXPLICIT-REFS

The following is an extension of REC.

3.2.1 Concrete Syntax

Examples of expressions in EXPLICIT-REFS:

```
newref(2)
   let a=newref(2)
   in a
  let a=newref(2)
   in deref(a)
   let a=newref(2)
   in setref(a,deref(a)+1)
   let a=newref(2)
   in begin
        setref(a,deref(a)+1);
14
        deref(a)
      end
16
   let g =
18
        let counter = newref(0)
        in proc (d) {
            begin
             setref(counter, deref(counter)+1);
22
             deref(counter)
            end
```

```
}
26 in (g 11) - (g 22)
```

```
\langle Expression \rangle ::= \langle Number \rangle
\langle Expression \rangle ::= \langle Identifier \rangle
                      ::= \langle Expression \rangle \langle BOp \rangle \langle Expression \rangle
(Expression)
\langle \mathsf{Expression} \rangle ::= \mathsf{zero}?(\langle \mathsf{Expression} \rangle)
\langle Expression \rangle ::= if \langle Expression \rangle then \langle Expression \rangle else \langle Expression \rangle
(Expression)
                      ::= let (Identifier) = (Expression) in (Expression)
(Expression)
                      ::= (\langle \mathsf{Expression} \rangle)
\langle \mathsf{Expression} \rangle ::= \mathsf{proc}(\langle \mathsf{Identifier} \rangle) \{\langle \mathsf{Expression} \rangle\}
\langle Expression \rangle ::= (\langle Expression \rangle \langle Expression \rangle)
\langle Expression \rangle ::= letrec \langle Identifier \rangle (\langle Identifier \rangle) = \langle Expression \rangle in \langle Expression \rangle
(Expression)
                      ::= newref((Expression))
                      ::= deref((Expression))
(Expression)
(Expression)
                      ::= setref((Expression), (Expression))
                      ::= begin \langle Expression \rangle^{*(;)} end
(Expression)
\langle BOp \rangle
                       ::= + | - | * | /
```

The notation *(;) above the nonterminal $\langle Expression \rangle$ in the production for begin/end indicates zero or more expressions separated by semi-colons.

3.2.2 Abstract Syntax

```
type expr =
      Var of string
      Int of int
       Add of expr*expr
       Sub of expr*expr
       Mul of expr*expr
6
     Div of expr*expr
     Let of string*expr*expr
     | IsZero of expr
     ITE of expr*expr*expr
10
       Proc of string*expr
     App of expr*expr
12
     Letrec of string*string*expr*expr
      NewRef of expr
       DeRef of expr
       SetRef of expr*expr
       BeginEnd of expr list
     Debug of expr
```

3.2.3 Interpreter

3.2.3.1 Specification

We assume given a set of (symbolic) memory locations \mathbb{L} . We write ℓ, ℓ_i for memory locations. A heap or **store** is a partial function from memory locations to expressed values. The set of

$$\frac{\mathsf{e}, \rho, \sigma \Downarrow v, \sigma' \quad \ell \not\in \mathsf{dom}(\sigma')}{\mathsf{NewRef}\,(\mathsf{e}), \rho, \sigma \Downarrow \ell, \sigma' \oplus \{\ell := v\}} \, \mathsf{ENewRef}$$

$$\frac{\mathsf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \in \mathsf{dom}(\sigma')}{\mathsf{DeRef}\,(\mathsf{e}), \rho, \sigma \Downarrow \sigma'(v), \sigma'} \, \mathsf{EDeRef}$$

$$\frac{\mathsf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \not\in \mathbb{L}}{\mathsf{DeRef}\,(\mathsf{e}), \rho, \sigma \Downarrow \mathit{error}, \sigma'} \, \mathsf{EDeRefErr1} \quad \frac{\mathsf{e}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad v \not\in \mathsf{dom}(\sigma')}{\mathsf{DeRef}\,(\mathsf{e}), \rho, \sigma \Downarrow \mathit{error}, \sigma'} \, \mathsf{EDeRefErr2}$$

$$\frac{\mathsf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \in \mathbb{L} \quad \mathsf{e2}, \rho, \sigma' \Downarrow w, \sigma''}{\mathsf{SetRef}\,(\mathsf{e1}, \mathsf{e2}), \rho, \sigma \Downarrow \mathit{unit}, \sigma'' \oplus \{v := w\}} \, \mathsf{ESetRef} \quad \frac{\mathsf{e1}, \rho, \sigma \Downarrow v, \sigma' \quad v \not\in \mathbb{L}}{\mathsf{SetRef}\,(\mathsf{e1}, \mathsf{e2}), \rho, \sigma \Downarrow \mathit{error}, \sigma'} \, \mathsf{ESetRefErr}$$

$$\frac{\mathsf{n} > 0 \quad (\mathsf{ei}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1...n}}{\mathsf{BeginEnd}([\mathsf{e1}; \dots; \mathsf{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \, \mathsf{EBeginEndNE}$$

Figure 3.1: Evaluation rules for EXPLICIT-REFS (error propagation rules omitted)

stores is denoted S:

$$\mathbb{S} := \mathbb{L} \rightharpoonup \mathbb{EV}$$

where the set of expressed values includes locations:

$$\mathbb{E}\mathbb{V}:=\mathbb{Z}\cup\mathbb{B}\cup\mathbb{U}\cup\mathbb{C}\mathbb{L}\cup\mathbb{L}$$

Also among expressed values we find $\mathbb{U} := \{unit\}$. This new value will be explained below, when we describe the evaluation rules for EXPLICIT-REFS.

Evaluation judgements in EXPLICIT-REFS take the following form, where e is an expression, ρ and environment, σ the initial store, r the result and σ' the final store

$$e, \rho, \sigma \Downarrow r, \sigma'$$

Note that the result of evaluating an expression now returns both a result and an updated store. The evaluation rules for EXPLICIT-REFS are given in Figure 3.1. The rule ESetRef and ESetRefErr describe the behavior of assignment. Notice that an assignment such as SetRef(e1,e2) is evaluated to <u>cause an effect</u>, namely update the contents of the location obtained from evaluating e1 with the value obtained from evaluating e2. We do not expect to get any meaningful value back. However, all expressions have to denote a value. As a consequence, we use a new expressed value unit, as the expressed value returned by an assignment.

3.2.3.2 Implementing Stores

The implementation of the evaluator for EXPLICIT-REFS requires that we first implement stores. Since a store is a mutable data structure we will use OCaml arrays. The following interface file declares the types of the values in the public interface of the store. These values include a parametric type constructor Store.t, the type of the store itself and multiple functions.

```
open Ds
type 'a t

val empty_store : int -> 'a -> 'a t
val get_size : 'a t -> int
val new_ref : 'a t -> int -> 'a -> int
val deref: 'a t -> int -> 'a ea_result
val set_ref : 'a t -> int -> 'a -> unit ea_result
val string_of_store : ('a -> string) -> 'a t -> string

store.mli
```

These operations are:

- empty_store n v returns a store of size n where each element is initialized to v
- get_size s returns the number of elements in the store.
- new ref s v stores v in a fresh location and returns the location.
- deref s 1 returns the contents of location 1, prefixed by Some, in the store s. This operation fails, returning None, if the location is out of bounds.
- set_ref s 1 v updates the contents of 1 in s with v. It fails, returning None, if the index is out of bounds.
- string_of_store to_str s returns a string representation of s resulting from applying to_str to each element.

Each of the above operations implemented in store.ml.

```
open Ds
2
   type 'a t = { mutable data: 'a array; mutable size: int}
    (* data is declared mutable so the store may be resized *)
   let empty_store : int -> 'a -> 'a t =
    fun i v -> { data=Array.make i v; size=0 }
   let get_size : 'a t -> int =
    fun st -> st.size
10
  let enlarge_store : 'a t -> 'a -> unit =
12
    fun st v ->
    let new_array = Array.make (st.size*2) v
    in Array.blit st.data 0 new_array 0 st.size;
    st.data<-new_array
16
  let new_ref : 'a t -> 'a -> int =
   fun st v ->
```

```
if Array.length (st.data)=st.size
     then enlarge_store st v
     else ();
     begin
       st.data.(st.size) <-v;
       st.size<-st.size+1;</pre>
      st.size-1
26
28
  let deref : 'a t -> int -> 'a ea_result =
     fun st 1 ->
     if l>=st.size
32
     then error "Index out of bounds"
     else return (st.data.(1))
  let set_ref : 'a t -> int -> 'a -> unit ea_result =
     fun st 1 v ->
     if 1>=st.size
     then error "Index out of bounds"
     else return (st.data.(1)<-v)
   let rec take n = function
    | [] -> []
    x::xs when n>0 -> x::take (n-1) xs
    | _ -> []
   let string_of_store' f st =
   let ss = List.mapi (fun i x -> string_of_int i^"->"^f x) @@ take st.size @@ Array.to_list st.
48
     String.concat ",\n" ss
50
  let string_of_store f st =
     match st.size with
     | 0 -> ">>Store:\nEmpty"
     | _ -> ">>Store:\n"^ string_of_store' f st
                                                                            store.ml
```



In OCaml, every .ml file is wrapped into a module. Modules package together related definitions and help provide consistent namespaces. For example, store.ml will be wrapped in a module called Store. Modules can provide not just functions but also types and submodules, among others. By default, all definitions provided in a module are accessible. Through interface files one may restrict the definitions that are accessible. For example, the store.mli file above, lists the definitions that are to be made accessible within the module Store.

3.2.3.3 Implementation

We could now follow the ideas we developed for environments and have stores threaded for us behind the scenes. This would lead to a similar extension of our current result type ea_result so that it also abstracts over the store. Also, the updated store would have to be returned. Thus the return type result would have be updated to return a pair consisting of the updated store

and the result itself¹. However, in order to keep things simple and since the concept of threading behind the scenes has already been introduced via environments, we choose to hold the store in a top-level or global variable g_store .

```
let g_store = Store.empty_store 20 (NumVal 0) interp.ml
```

g_store denotes a store of size 20, whose values have arbitrarily been initialized to NumVal 0.

Next we consider the new expressed values, namely symbolic locations and unit. Locations will be denoted by an integer wrapped inside a RefVal constructor. For example, RefVal 7 is a pointer to memory location 7.

Next we move on to the interpreter, only addressing the new variants.

```
let rec eval_expr : expr -> exp_val ea_result =
     fun e ->
     match e with
     NewRef(e) ->
       eval_expr e >>= fun ev ->
       return (RefVal (Store.new_ref g_store ev))
     DeRef(e) ->
8
       eval_expr e >>=
       int_of_refVal >>= fun l ->
       Store.deref g_store 1
10
     SetRef(e1,e2)
       eval_expr e1 >>=
12
       int_of_refVal >>= fun l ->
       eval_expr e2 >>= fun ev ->
14
       Store.set_ref g_store l ev >>= fun _ ->
       return UnitVal
16
     | BeginEnd([]) ->
       return UnitVal
18
     | BeginEnd(es) ->
       sequence (List.map eval_expr es) >>= fun 1 ->
20
       return (List.hd (List.rev 1))
     | Debug(_e) ->
       string_of_env >>= fun str_env ->
       let str_store = Store.string_of_store string_of_expval g_store
       in (print_endline (str_env^"\n"^str_store);
       error "Debug called")
26
     | _ -> failwith ("Not implemented: "^string_of_expr e)
                                                                           interp.ml
```

¹This handling of the store in the background, including its auxiliary data types, is know as a <u>state monad</u>. Thus we would end up with a combination of error, reader, and state monads. Combining monads can be done through monad transformers.

3.2.4 Extended Example: Encoding Objects

EXPLICIT-REFS with records

```
letrec self(s) =
              { inc = proc (d) { setref(s,deref(s)+d) };
                read = proc (x) { deref(s) };
                reset = proc (d) {
                        let current = ((self s).read 0)
                        in ((self (s)).inc (-current))}
6
              }
   in let new_counter = proc(init) {
                           let s = newref(init)
                           in (self s)
10
   in let c= (new_counter 0)
   in begin
       (c.inc 1);
       (c.inc 2);
16
       (c.reset 0);
       (c.read 0)
```

3.3 IMPLICIT-REFS

The following is an extension of REC.

3.3.1 Concrete Syntax

Examples of expressions in IMPLICIT-REFS

```
y

12 end

14 let x=2

in let f = proc (n) { begin set x=x+1; 1 end }

16 in let g = proc (n) { begin set x=x+1; 2 end }

in begin

(f 0)+(g 0);

x

20 end
```

```
\langle Expression \rangle ::=
                              (Number)
(Expression)
                              (Identifier)
                     ::=
\langle Expression \rangle ::= \langle Expression \rangle \langle BOp \rangle \langle Expression \rangle
\langle Expression \rangle ::= zero?(\langle Expression \rangle)
\langle Expression \rangle ::= if \langle Expression \rangle then \langle Expression \rangle else \langle Expression \rangle
                     ::= let (Identifier) = (Expression) in (Expression)
(Expression)
\langle \mathsf{Expression} \rangle ::= (\langle \mathsf{Expression} \rangle)
\langle \mathsf{Expression} \rangle ::= \mathsf{proc}(\langle \mathsf{Identifier} \rangle) \{\langle \mathsf{Expression} \rangle \}
                     ::= (\langle Expression \rangle \langle Expression \rangle)
(Expression)
(Expression)
                     ::= letrec(Identifier)((Identifier))=(Expression)in(Expression)
                     ::= set (Identifier) = (Expression)
(Expression)
                     ::= begin \langle Expression \rangle^{+(i)} end
⟨Expression⟩
\langle BOp \rangle
                      ::= + | - | * | /
```

3.3.2 Abstract Syntax

```
type expr =
      Var of string
       Int of int
       Add of expr*expr
       Sub of expr*expr
     | Mul of expr*expr
       Div of expr*expr
     Let of string*expr*expr
8
       IsZero of expr
     ITE of expr*expr*expr
10
     | Proc of string*expr
     App of expr*expr
12
       Letrec of string*string*expr*expr
       Set of string*expr
       BeginEnd of expr list
     Debug of expr
```

3.3.3 Interpreter

3.3.3.1 Specification

Since in IMPLICIT-REFS all identifiers are mutable, the environment will map all identifiers to locations in the store. Evaluation judgements in IMPLICIT-REFS take the following form, where

$$\frac{\sigma(\rho(\mathrm{id})) = v}{\mathrm{Var}(\mathrm{id}), \rho, \sigma \Downarrow v, \sigma} \, \mathrm{EVar} \quad \frac{\rho(\mathrm{id}) \notin \mathbb{L} \, \mathrm{or} \rho(\mathrm{id}) \notin \mathrm{dom}(\sigma)}{\mathrm{Var}(\mathrm{id}), \rho, \sigma \Downarrow error, \sigma} \, \mathrm{EVarErr}$$

$$\frac{\mathrm{e1}, \rho, \sigma_1 \Downarrow w, \sigma_2}{\mathrm{Let}(\mathrm{id}, \mathrm{e1}, \mathrm{e2}), \rho, \sigma_1 \Downarrow v, \sigma_3} \, \mathrm{ELet}$$

$$\frac{\mathrm{Elt}(\mathrm{id}, \mathrm{e1}, \mathrm{e2}), \rho, \sigma_1 \Downarrow v, \sigma_3}{\mathrm{Proc}(\mathrm{id}, \mathrm{e}), \rho, \sigma \Downarrow (\mathrm{id}, \mathrm{e}, \rho), \sigma} \, \mathrm{EProc}$$

$$\frac{\mathrm{e1}, \rho, \sigma \Downarrow (\mathrm{id}, \mathrm{e}, \tau), \sigma_1 \quad \mathrm{e2}, \rho, \sigma_1 \Downarrow w, \sigma_2 \quad \ell \notin \mathrm{dom}(\sigma_2) \quad \mathrm{e}, \tau \oplus \{\mathrm{id} := \ell\}, \sigma_2 \oplus \{\ell := w\} \Downarrow v, \sigma_3}{\mathrm{App}(\mathrm{e1}, \mathrm{e2}), \rho, \sigma \Downarrow v, \sigma_3} \, \mathrm{EApp}$$

$$\frac{\mathrm{e}, \rho, \sigma \Downarrow v, \sigma'}{\mathrm{Set}(\mathrm{id}, \mathrm{e}), \rho, \sigma \Downarrow unit, \sigma' \oplus \{\rho(\mathrm{id}) := v\}} \, \mathrm{ESet}$$

$$\frac{\rho(\mathrm{id}) \notin \mathbb{L} \, \mathrm{or} \, \rho(\mathrm{id}) \notin \mathrm{dom}(\sigma)}{\mathrm{Set}(\mathrm{id}, \mathrm{e}), \rho, \sigma \Downarrow error, \sigma} \, \mathrm{ESetErr}$$

$$\frac{\mathrm{n} > 0 \quad (\mathrm{ei}, \rho, \sigma_i \Downarrow v_i, \sigma_{i+1})_{i \in 1...n}}{\mathrm{BeginEnd}([\mathrm{e1}; \dots; \mathrm{en}]), \rho, \sigma_1 \Downarrow v_n, \sigma_{n+1}} \, \mathrm{EBeginEndNE}$$

$$\frac{\mathrm{BeginEnd}([\mathrm{e1}; \dots; \mathrm{en}]), \rho, \sigma \Downarrow unit, \sigma}{\mathrm{BeginEndE}} \, \mathrm{EBeginEndE}$$

Figure 3.2: Evaluation rules for IMPLICIT-REFS (error propagation rules are omitted)

e is an expression, ρ and environment, σ the initial store, r the result and σ' the final store

$$e, \rho, \sigma \downarrow r, \sigma'$$

As mentioned, ρ maps identifiers to locations, it **no longer** maps them to expressed values. Hence identifier lookup now has to lookup the location first in the environment and then access the contents in the store. This is exactly what the rule EVar states:

$$\frac{\sigma(\rho(\mathrm{id})) = v}{\mathrm{Var}(\mathrm{id}), \rho, \sigma \Downarrow v, \sigma} \, \mathrm{EVar}$$

Indeed, $\rho(id)$ denotes a location whose contents is looked up in the store σ . If $\rho(id)$ is not a valid location, then an error is returned, as described by rule EVarErr. The full set of evaluation rules are given in Figure 3.2.

3.3.3.2 Implementation

We address the implementation of the evaluator. For now we ignore Letrec and then take it up later. Instead we focus on the App(e1,e2) case, which needs some minor updating, and also on

the new variants.

Regarding the App(e1,e2) case, we need to slightly modify the apply_clos function. We briefly recall the code for apply_clos as implemented in the PROC (Figure 2.8):

```
let rec apply_clos : string*Ast.expr*env -> exp_val -> exp_val ea_result =
fun (id,e,en) ev ->
return en >>+
(extend_env id ev >>+
eval_expr e)
```

Note that evaluation of the body requires extending the environment with a new key-value pair, namely (id,ev), where ev is the expressed value supplied as argument. Environments no longer map identifiers to expressed values, but to locations. So we first need to allocate ev in the store in a fresh location 1 and then extend the environment with the key-value pair (id,1). The updated code for apply_clos is given below.

```
let rec apply_clos : string*Ast.expr*env -> exp_val -> exp_val ea_result =
fun (id,e,en) ev ->
return en >>+
  (extend_env id (RefVal (Store.new_ref g_store ev)) >>+
eval_expr e)
```

We now address the new cases (and the ones we need to modify) for the interpreter:

```
| Var(id) ->
       apply_env id >>=
       int_of_refVal >>=
                            (* make sure id is mapped to a location *)
       Store.deref g_store (* if so, dereference it *)
     Let(v,def,body) ->
       eval_expr def >>= fun ev ->
                                         (* evaluate definition *)
       let 1 = Store.new_ref g_store ev (* allocate it in the store *)
       in extend_env v (RefVal 1) >>+
                                        (* extend env with new key-value pair *)
9
                                         (* eval body in extended env *)
       eval_expr body
     | Set(id,e) ->
11
       eval_expr e >>= fun ev -> (* eval RHS *)
       apply_env id >>=
13
       int_of_refVal >>= fun 1 -> (* make sure id is mapped to location *)
       Store.set_ref g_store l ev >>= fun _ -> (* update the store *)
       return UnitVal
     | BeginEnd([]) ->
17
       return UnitVal
     | BeginEnd(es) ->
19
       sequence (List.map eval_expr es) >>= fun vs ->
       return (List.hd (List.rev vs))
21
    | Debug(_e) ->
       string_of_env >>= fun str_env ->
23
       let str_store = Store.string_of_store string_of_expval g_store
       in (print_endline (str_env^"\n"^str_store);
25
       error "Debug called")
     | _ -> failwith ("Not implemented: "^string_of_expr e)
```

3.3.3.3 letrec Revisited

Our implementation of letrec in REC consisted in adding a specific entry in the environment to signal the declaration of a recursive function. Then, upon lookup, a closure was created on the

fly. This is the code from REC. The highlighted excerpt Ok (ProcVal (par,body,env)) indicates that a closure is being created.

```
type env =
      EmptyEnv
     ExtendEnv of string*exp_val*env
     | ExtendEnvRec of string*string*Ast.expr*env
   let rec apply_env : string -> exp_val ea_result =
     fun id env -
     match env with
     | EmptyEnv -> Error (id^" not found!")
     ExtendEnv(v,ev,tail) ->
10
       if id=v
       then Ok ev
12
       else apply_env id tail
     ExtendEnvRec(v,par,body,tail) ->
14
       if id=v
       then Ok (ProcVal (par,body,env))
16
       else apply_env id tail
                                                                               ds.ml
```

We could follow the same approach in IMPLICIT-REFS. But there is a better way, which avoids having to create closures on the fly. The idea is to allow circular environments. That is, an environment $_{\tt env}$ that has an entry to a location on the store that holds a closure whose environment has a reference to this same location.

So we first remove the special entry in environments for letrec declarations since they will no longer be needed:

```
type env =
      EmptyEnv
     ExtendEnv of string*exp_val*env
     ExtendEnvRec of string*string*Ast.expr*env
   let rec apply_env : string -> exp_val ea_result =
     fun id env ->
     match env with
     | EmptyEnv -> Error (id^" not found!")
     ExtendEnv(v,ev,tail) ->
10
       if id=v
       then Ok ev
12
       else apply_env id tail
     + ExtendEnvRec(v,par,body,tail) ->
14
       if id=v
       then Ok (ProcVal (par, body, env))
16
       else apply_env id tail
                                                                                   ds.ml
```

We now use "back-patching" to code the circular environment:

```
let rec eval_expr : expr -> exp_val ea_result =
   fun e ->
match e with
   Letrec(id,par,e,target) ->
let l = Store.new_ref g_store UnitVal in
   extend_env id (RefVal l) >>+
   (lookup_env >>= fun env ->
```



Parenthesis right after (>>+) are necessary since (>>=) and (>>+) are left-associative. Remove them, execute the resulting interpreter on an example expression and explain what goes wrong.

3.4 Parameter Passing Methods

We consider several parameter passing methods in IMPLICIT-REFS.

3.4.1 Call-by-Value

This method consists in first evaluating the argument, before passing on its value to the function. This is the parameter passing method we have implemented in PROC and all the languages that extend it.

3.4.2 Call-by-Reference

If the argument to a function is a variable, then we provide a copy of its reference to the function. Otherwise, the argument is processed just like in call-by-value. For example, evaluation of

returns Ok (NumVal 2) in IMPLICIT-REFS. However, using call-by-reference, it will return Ok (NumVal 3). It is helpful to place a breakpoint inside the body of the f, evaluate the resulting expression and examine the environment and store.

3.4.2.1 Modifying the Interpreter

```
let rec value_of_operand : expr -> exp_val ea_result =
  fun e ->
  match e with

4  | Var(id) -> apply_env id
  | _ -> eval_expr e >>= fun ev ->
    return (RefVal (Store.new_ref g_store ev))
and

8  apply_clos =
  ...

10 and
  eval_expr : expr -> exp_val ea_result =
  fun e ->
  match e with
```

The following example shows how one may swap the contents of two variables:

Returns Ok (NumVal 1).

3.4.3 Call-by-Name

Consider the following expression. What is the result of its evaluation?

```
letrec infinite_loop (x) = (infinite_loop (x+1))
in let f = proc (y) { 11 }
in (f (infinite_loop 0))
```

The parameter z is not used. However, the argument to f, namely (infinite_loop 0), is evaluated all the same. The call-by-name parameter passing method consists in freezing the evaluation of arguments until they are actually needed. This is achieved as follows. In an application App(e1,e2), if e2 is an identifier, then call-by-name proceeds just like call-by-reference: it passes a copy of the address of the identifier, after looking it up in the environment, as an argument to the parameter. Regardless of whether the value of e2 will be used, copying an address is a constant time operation, hence not costly. However, if e2 is an expression different from an identifier, then its evaluation does not take place. Rather, e2 together with the current environment are stored for later evaluation. The pair consisting of e2 and the current environment is called a thunk. A thunk is just a closure without the formal parameter. We next implement call-by-name, the parameter passing method that implements this idea. We will use the code for call-by-reference as a starting point.

We begin by adding thunks to the set of expressed values:

ds.ml

Note that we use Thunk rather than ThunkVal as the name of the constructor, to emphasize that a thunk is not a run-time value that may be returned as the result of a evaluation. Next we update value_of_operand. If the argument or operand is a variable, our interpreter behaves just like in call-by-reference. However, if it is not a variable, we create a thunk:

```
let rec value_of_operand =
fun op ->
match op with

| Var id -> apply_env id
| _ ->
lookup_env >>= fun en ->
return (RefVal (Store.new_ref g_store (Thunk(op, en))))
interp.ml
```

Finally, we have to consider what happens when we lookup the contents of the store through the reference assigned to a variable in the environment and it consists of a thunk Thunk(e,en). We "thaw" the thunk by evaluating e under the environment en:

```
let rec eval_expr : expr -> exp_val ea_result =
    fun e ->

match e with
    | Int(n) -> return (NumVal n)

Var(id) ->
    apply_env id >>=

int_of_refVal >>=
    Store.deref g_store >>= fun ev ->

(match ev with
    | Thunk(e,en) -> return en >>+ eval_expr e

    | _ -> return ev)
    ...

interp.ml
```

Evaluate the example from the beginning of this section in CBN. Then do the same but this time inserting a breakpoint in the body of f:

```
letrec infinite_loop (x) = (infinite_loop (x+1))
in let f = proc (y) { debug(11) }
in (f (infinite_loop 0))
```

3.4.4 Call-by-Need

One drawback of call-by-name is that a thunk is "thawed" every time it is needed. Consider the following example:

```
letrec f(x) = if zero?(x) then 1 else x*(f (x-1))
in let g = proc (y) { y+y+y+y }
in (g (f 5))
```

Here the factorial of 5 is computed four times, one for each occurrence of y in g. A more reasonable approach is to do this once, the first time, and then store the result for further uses. This optimization technique is called memoization. Call-by-need consists in applying this

optimization technique to call-by-name. All we need to do is update the implementation for the variable constructor in eval_expr:

```
let rec eval_expr : expr -> exp_val ea_result =
     fun e ->
     match e with
     | Int(n) -> return @@ NumVal n
     | Var(id) ->
       apply_env id >>=
       int_of_refVal >>= fun 1 ->
       Store.deref g_store 1 >>= fun ev ->
      (match ev with
        Thunk(e,en) ->
          return en >>+
          eval_expr e >>= fun ev ->
          Store.set_ref g_store 1 ev >>= fun _ ->
         return ev
       _ -> return ev)
15
                                                                         interp.ml
```

The factorial example above now involves computing factorial of 5 just once.

Call-by-need and call-by-name may return different results for effectful computation. Evaluate the following expression in CBN and CBNeed:

3.5 Exercises

Exercise 3.5.1. Depict the environment and store at the breakpoint for the following EXPLICIT-REFS programs:

```
in debug(a)
```

Exercise 3.5.2. Consider the following extension of LET with records (Exercise 2.2.6). It has the same syntax except that one can declare a field to be mutable by using <= instead of =. For example, the ssn field is immutable but the age field is mutable; age is then updated to 31:

```
let p = {ssn = 10; age <= 30}
in begin
    p.age <= 31;
    p.age
end</pre>
```

Evaluating this expression should produce 0k (NumVal 31). This other expression should produce 0k (RecordVal [("ssn", (false, NumVal 10)); ("age", (true, RefVal 1))]):

```
let p = {ssn = 10; age <= 30}
in begin
    p.age <= 31;
    p
end</pre>
```

Updating an immutable field should not be allowed. For example, the following expression should report an error "Field not mutable":

```
let p = { ssn = 10; age = 20}
in begin
   p.age <= 21;
   p.age
end</pre>
```

The abstract syntax requires modifying the Record constructor and adding a new one for field update:

```
type expr =
...
/ Record of |(string*(bool*expr))| list
/ SetField of expr*string*expr
```

For example,

```
# parse "
let p = {ssn = 10; age <= 30}
in begin
        p.age <= 31;
        p
        end";;
- : expr =
Let ("p", Record [("ssn", (false, Int 10)); ("age", (true, Int 30))],
        BeginEnd [SetField (Var "p", "age", Int 31); Var "p"])</pre>
utop
```

Here false indicates the field is immutable and true that it is mutable. You are asked to implement the interpreter extension. The RecordVal constructor has been updated for you.

As for <code>eval_expr</code>, the case for <code>Record</code> has already been updated for you. You are asked to update <code>Proj</code> and complete <code>SetField</code>:

```
let rec eval_expr : expr -> exp_val ea_result = fun e ->
match e with
| Record(fs) ->
sequence (List.map process_field fs) >>= fun evs ->
return (RecordVal (addIds fs evs))
| Proj(e,id) ->
error "update"
| SetField(e1,id,e2) ->
error "implement"
and

process_field (_id,(is_mutable,e)) =
eval_expr e >>= fun ev ->
if is_mutable
then return (RefVal (Store.new_ref g_store ev))
else return ev
```

Exercise 3.5.3. Depict the environment and store extant at the breakpoint in the following IMPLICIT-REFS expressions.

```
1. \frac{1}{let \ a = 2}
   in let b = 3
   in begin
       set \ a = b;
       debug(a)
   let a = 2
   in let b = a
   in begin
       set b = 3;
       debug(a)
      end
3.
   let a = 2
   in let b = proc(x) {
                  begin
                   set a = x;
                   debug(a)
              }
   in (b 3)
```

```
4. 
let a = 2
in let b = proc(x) {
    begin
    set a = x;
    a
    end
}
in (b 3) + debug((b 4))
```

Exercise 3.5.4. Depict the environment and store at the breakpoint first assuming call-by-reference as parameter passing method and then call-by-value.

```
1.

let a=1

in let b=2

in let f = proc (x) { proc (d) { debug(set x=x+d) }}

in ((f a) (b+1))
```

Chapter 4

Types

This chapter extends the REC language to support type-checking.

4.1 CHECKED

4.1.1 Concrete Syntax

```
\langle Expression \rangle ::=
                                   (Number)
\langle Expression \rangle ::= \langle Identifier \rangle
\langle Expression \rangle ::= \langle Expression \rangle \langle BOp \rangle \langle Expression \rangle
\langle Expression \rangle ::= zero?(\langle Expression \rangle)
\langle Expression \rangle ::= if \langle Expression \rangle then \langle Expression \rangle else \langle Expression \rangle
\langle Expression \rangle ::= let \langle Identifier \rangle = \langle Expression \rangle in \langle Expression \rangle
\langle Expression \rangle ::= (\langle Expression \rangle)
\langle Expression \rangle ::= proc(\langle Identifier \rangle : \langle Type \rangle) \{\langle Expression \rangle \}
\langle \mathsf{Expression} \rangle ::= (\langle \mathsf{Expression} \rangle \langle \mathsf{Expression} \rangle)
(Expression)
                         ::= letrec\langle Identifier \rangle (\langle Identifier \rangle : \langle Type \rangle) : \langle Type \rangle = \langle Expression \rangle in \langle Expression \rangle
\langle \mathsf{BOp} \rangle
                          ::= + | - | * | /
⟨Type⟩
                          ::= int
⟨Type⟩
                          ::= bool
⟨Type⟩
                          ::= \ \langle \mathsf{Type} \rangle \text{--} \langle \mathsf{Type} \rangle
⟨Type⟩
                          ::= (\langle \mathsf{Type} \rangle)
```

4.1.2 Abstract Syntax

```
type expr =
2   | Var of string
   | Int of int
4   | Sub of expr*expr
   | Let of string*expr*expr
```

4.1.3 Type-Checker

We specify the behavior of our type-checker before implementing it, much like we do with interpreters. For this task we use **type systems**. A type system is an inductive set that helps identify a subset of the expressions that are considered to be <u>well-typed</u>. The elements of the inductive set are called **type judgements**. Which type judgements belong to the set and which don't is determined by a sert of <u>type rules</u>. A type judgement is an expression of the form

where Γ is a type environment, e is an expression in CHECKED, and t is a type expression. These components together with the type rules are introduced below.

4.1.3.1 Specification

As mentioned, a type judgement consists of a of type environment, an expression in CHECKED and a type. Types are defined as follows:

$$t ::= \inf|\operatorname{bool}|t \to t$$

A **type context** is a partial function that assigns a type to an identifier. Type contexts are required for typing expressions that contain free variables. For example, an expression such as x+2 will require that we have the type of x at our disposal in order to determine whether x+2 is typable at all. If the type of x were bool, then it is not typable; but if the type of x is int, then it is. Type contexts are defined as follows:

$$\Gamma ::= \epsilon \mid \Gamma, id : t$$

We use ϵ to denote the empty type environment. Also, $\Gamma, id: t$ assigns type t to identifier id and behaves as Γ for identifiers different from id. We assume that Γ does not have repeated entries for the same identifier. An example of a type contexts is $\epsilon, x: \mathtt{int}, y: \mathtt{bool}$. We abbreviate it as $x: \mathtt{int}, y: \mathtt{bool}$.

The type rules are given in Figure 4.1.

4.1.3.2 Towards and Implementation

Our type checker will behave very much like our interpreter, except that instead of manipulating runtime values such as integers and booleans, it manipulates types like int and bool. One might say that a type checker is a <u>symbolic</u> evaluator, where our symbolic values are the types. This analogy allows us to apply the ideas we have developed on well-structuring an evaluator to our type checker. Thus one might be tempted to state the type of our type-checker as

$$\begin{array}{lll} \hline \Gamma \vdash \mathbf{n} : \mathbf{int} & \frac{\Gamma(\mathbf{x}) = t}{\Gamma \vdash \mathbf{x} : t} \, \mathsf{TVar} & \frac{\Gamma \vdash \mathbf{e} : \mathbf{int}}{\Gamma \vdash \mathsf{zero}?(\mathbf{e}) : \mathsf{bool}} \, \mathsf{TIsZero} \\ & \frac{\Gamma \vdash \mathbf{e}1 : \mathbf{int} & \Gamma \vdash \mathbf{e}2 : \mathbf{int}}{\Gamma \vdash \mathbf{e}1 : \mathbf{e}1 : \mathbf{e}2 : \mathbf{int}} \, \mathsf{TSub} \\ & \frac{\Gamma \vdash \mathbf{e}1 : \mathsf{bool} & \Gamma \vdash \mathbf{e}2 : t & \Gamma \vdash \mathbf{e}3 : t}{\Gamma \vdash \mathsf{if} \, \mathbf{e}1 \, \mathsf{then} \, \mathbf{e}2 \, \mathsf{else} \, \mathbf{e}3 : t} \, \mathsf{TITE} \\ & \frac{\Gamma \vdash \mathbf{e}1 : t1 & \Gamma, \mathsf{id} : t1 \vdash \mathbf{e}2 : t2}{\Gamma \vdash \mathsf{let} \, \mathsf{id}=\mathsf{e}1 \, \mathsf{in} \, \mathbf{e}2 : t2} \, \mathsf{TLet} \\ & \frac{\Gamma \vdash \mathsf{rator} : t1 \to t2 & \Gamma \vdash \mathsf{rand} : t1}{\Gamma \vdash \mathsf{(rator} \, \mathsf{rand)} : t2} \, \mathsf{TApp} \\ & \frac{\Gamma, \mathsf{id} : t1 \vdash \mathbf{e} : t2}{\Gamma \vdash \mathsf{proc} \, (\mathsf{id}:\mathsf{t1}) \, \{\mathbf{e}\} : t1 \to t2} \, \mathsf{TProc} \end{array}$$

Figure 4.1: Type Rules for CHECKED

```
chk_expr : expr -> texpr ea_result
```

reflecting that, given an expression, it returns a function that given a type environment returns either a type or an error. Note, however, that ea_result abstracts environments, and not type environments:

```
type 'a ea_result = env -> 'a result
```

We could create a new type constructor, let us call it tea_result, where env is replaced with tenv:

```
type 'a tea_result = tenv -> 'a result
```

But we would also have to duplicate all of return, error, (>>=), lookup_env, etc. to support this new type and end up having two copies of all these operations (one supporting ea_result and one supporting tea_result) with the exact same code. Since the only difference between ea_result and tea_result is the kind of environment they abstract over, we choose to define a more general type constructor a_result ("a" for "abstracted") and have both of these be instances of them:

```
type ('a,'b) a_result = 'b -> 'a result
```

Notice that, contrary to ea_result and tea_result, the type constructor a_result is parameterized over two types,

- 1. the type a representing the result of the computation, and
- 2. the type b representing that over which the function type is being abstracted over.

```
type 'a result = Ok of 'a | Error of string
   type ('a,'b) a_result = 'b -> 'a result
   let return : 'a -> ('a,'b) a_result =
    fun v ->
     fun env -> 0k v
   let error : string -> ('a,'b) a_result =
    fun s ->
10
    fun env -> Error s
12
   let (>>=) : ('a,'c) a_result -> ('a -> ('b,'c) a_result) -> ('b,'c) a_result =
    fun c f ->
14
     fun env ->
     match c env with
16
     Error err -> Error err
     Ok v -> f v env
18
  let (>>+) : ('b,'b) a_result -> ('a,'b) a_result -> ('a,'b) a_result =
    fun c d ->
    fun env ->
     match c env with
     | Error err -> Error err
     Ok newenv -> d newenv
                                                                             reM.ml
```

Figure 4.2: The a_result type

The type ('a, 'b) a_result, together with its supporting operations are declared in Figure 4.2. Notice that the code for the supporting operations, namely return, error, (>>=) and (>>+) is exactly the same as before. The only difference is their type. That being said, we still call the formal parameter of type 'b in these operations, env.

With the newly declared type constructor a_result in place, we can now redefine ea_result and tea_result as instances of it. Indeed, ea_result is simply defined as:

4.1.3.3 Implementation

We next address the implementation of the type-checker for CHECKED. The code is given in Figure 4.3.

```
utop # chk "
let add = proc (x:int) { proc (y:int) { x+y }} in (add 1)";;

- : texpr ReM.result = Ok (FuncType (IntType, IntType))
utop
```

```
let rec chk_expr : expr -> texpr tea_result =
     fun e ->
     match e with
     | Int _n -> return IntType
     | Var id -> apply_tenv id
     | IsZero(e) ->
6
       chk_expr e >>= fun t ->
8
       if t=IntType
       then return BoolType
       else error "isZero: expected argument of type int"
10
     | Add(e1,e2) | Sub(e1,e2) | Mul(e1,e2) | Div(e1,e2) ->
12
       chk_expr e1 >>= fun t1 ->
       chk_expr e2 >>= fun t2 ->
       if (t1=IntType && t2=IntType)
14
       then return IntType
       else error "arith: arguments must be ints"
16
     | ITE(e1,e2,e3) ->
       chk_expr e1 >>= fun t1 ->
18
       chk_expr e2 >>= fun t2 ->
       chk_expr e3 >>= fun t3 ->
20
       if (t1=BoolType && t2=t3)
       then return t2
       else error "ITE: condition not bool/types of then-else do not match"
     Let(id,e,body) ->
       chk_expr e >>= fun t ->
       extend_tenv id t >>+
26
       chk_expr body
     | Proc(var,t1,e) ->
28
       extend_tenv var t1 >>+
       chk_expr e >>= fun t2 ->
30
       return (FuncType(t1,t2))
32
     App(e1,e2) ->
       chk_expr e1 >>=
       pair_of_funcType "app: " >>= fun (t1,t2) ->
       chk_expr e2 >>= fun t3 ->
       if t1=t3
36
       then return t2
       else error "app: type of argument incorrect"
38
     | Debug(_e) ->
       string_of_tenv >>= fun str ->
40
       print_endline str;
       error "Debug: reached breakpoint"
42
     | _ -> failwith "chk_expr: implement"
                                                                          checker.ml
```

Figure 4.3: Type checker for CHECKED

4.1.4 Adding Letrec

The typing rule for letrec is as follows:

```
\begin{array}{c} \Gamma, \texttt{par}: tVar, \texttt{id}: tPar \rightarrow tRes \vdash \texttt{body}: tRes \\ \Gamma, \texttt{id}: tPar \rightarrow tRes \vdash \texttt{target}: t \\ \hline \Gamma \vdash \texttt{letrec} \ \texttt{id} \ (\texttt{par}: \texttt{tPar}): \texttt{tRes} = \texttt{body} \ \texttt{in} \ \texttt{target}: t \end{array}
```

```
let rec chk_expr : expr -> texpr tea_result = function
| Letrec(id,param,tPar,tRes,body,target) ->
extend_tenv id (FuncType(tPar,tRes)) >>+
    (extend_tenv param tPar >>+
    chk_expr body >>= fun t ->
    if t=tRes
then chk_expr target
else error "LetRec: Type of rec. function does not match declaration")
```

4.1.5 Exercises

Exercise 4.1.1. Provide typing derivations for the following expressions:

```
    if zero?(8) then 1 else 2
    if zero?(8) then zero?(0) else zero?(1)
    proc (x:int) { x-2 }
    proc (x:int) { proc (y:bool) { if y then x else x-1 } }
    let x=3 in let y = 4 in x-y
    let two? = proc(x:int) { if zero?(x-2) then 0 else 1 } in (two? 3)
```

Exercise 4.1.2. Recall that an expression e is <u>typable</u>, if there exists a type environment Γ and a type expression t such that the typing judgement $\Gamma \vdash e : t$ is derivable. Argue that the expression x x (a variable applied to itself) is not typable.

Exercise 4.1.3. Give a typable term of each of the following types, justifying your result by showing a type derivation for that term.

```
    bool->int
    (bool -> int) -> int
    bool -> (bool -> bool)
    (s -> t) -> (s -> t), for any types s and t.
```

Exercise 4.1.4. Show that the following term is typable:

```
letrec\ int\ double\ (x:int) = if\ zero?(x) then\ 0 else\ (double\ (x-1))\ +\ 2 in\ double
```

CHAPTER 4. TYPES 73

Exercise 4.1.5. What is the result of evaluating the following expressions in CHECKED?

```
utop # check "
letrec int double (x:int) = if zero?(x)
                                else (double (x-1)) + 2
in (double 5)";;
                                                                              utop
utop # check "
letrec int double (x:int) = if zero?(x)
                                then 0
                                else (double (x-1)) + 2
in double";;
                                                                              utop
utop # check "
letrec bool double (x:int) = if zero?(x)
                                else -((double -(x,1)), -2)
in double"::
                                                                              utop
utop # check "
letrec bool double (x:int) = if zero?(x)
                                then 0
                                else 1
in double";;
                                                                              utop
utop # check "
letrec int double (x:bool) = if zero?(x)
                                then 0
                                else 1
in double";;
                                                                              utop
```

Exercise 4.1.6. Consider the extension of Exercise 2.2.4 where pairs are added to our language. In order to extend type-checking to pairs we first add pair types to the concrete syntax of types:

```
<Type> ::= int

<Type> ::= bool

<Type> ::= <Type> -> <Type>

<Type> ::= <Type> * <Type>

<Type> ::= (<Type>)
```

Recall from Exercise 2.2.4 that expressions are extended with a pair(e1,e2) construct to build new pairs and an unpair(x,y)=e1 in e2 construct that given an expression e1 that evaluates to a pair, binds e2 and e3 to the first and second component of the pair, respectively, in e3. Here are some examples of expressions in the extended language:

```
pair(3,4)
pair(pair(3,4),5)
pair(zero?(0),3)
pair(proc (x:int) { x-2 },4)
proc (z:int*int) { unpair (x,y)=z in x }
proc (z:int*bool) { unpair (x,y)=z in pair(y,x) }
```

You are asked to give typing rules for each of the two new constructs.

Exercise 4.1.7. Consider the following the extension of CHECKED with records, as introduced in Exercise 2.2.6. The concrete syntax for the new type constructor for records is given by the second to last production below:

```
<Type> ::= int

<Type> ::= bool

<Type> ::= <Type> -> <Type>

<Type> ::= \{ \langle \text{Identifier} \rangle = <Type>^{+(;)} \}

<Type> ::= (<Type>)
```

The abstract syntax is as follows:

```
type expr =
  / Var of string
  / Int of int
 | Sub of expr*expr
  / Let of string*expr*expr
  / IsZero of expr
  / ITE of expr*expr*expr
  / Proc of string*texpr*expr
  / App of expr*expr
  | Letrec of string*string*texpr*texpr*expr*expr
  Record of (string*expr) list
  | Proj of expr*string
and
texpr =
  / IntType
  / BoolType
  / FuncType of texpr*texpr
 RecordType of (string*texpr) list
```

For example,

- 1. {age=2; height=3} should have type {age:int; height:int}.
- 2. {age=2; present=zero?(0)} should have type {age:int; present:bool}.
- 3. $\{inc = proc(x:int) \{x+1\}; dec = proc(x:int) \{x-1\}\}$ should have type $\{inc:int->int; dec:int->int\}$.
- 4. $\{inc = proc(x:int) \{x+1\}; dec = proc(x:int) \{x-1\}\}.inc should have type int->int.$

CHAPTER 4. TYPES 75

- 5. {} should produce a type error since empty records are not allowed.
- 6. {age=2; height=3}.weight should produce a type error since there is no field named weight.

The additional typing rules are:

$$\frac{\Gamma \vdash e1:t1 \quad \dots \quad \Gamma \vdash en:tn \quad n>0 \quad li,i \in 1..n, \textit{distinct}}{\Gamma \vdash \{ \quad l1=e1; \quad \dots; \quad ln=en \}: \{l1:t1;\dots;ln:tn \}} \, \text{TRec}$$

$$\frac{\Gamma \vdash e:\{l1:t1;\dots;ln:tn\} \quad l=li, \; \textit{for some } i \in 1..n}{\Gamma \vdash e.\; l:ti} \, \text{TProj}$$

Extend the type checker chk_expr to deal with the two new constructs.

76 CHAPTER 4. TYPES

Chapter 5

Modules

5.1 Syntax

SIMPLE-MODULES is an extension to the EXPLICIT-REFS language. A program in SIMPLE-MODULES consists of a list of module declarations together with an expression (the "main" expression). Here is an example that consists of one module declaration, the module called m1, and a main expression consisting of a let expression. A module has an interface and a body.

```
module m1
2   interface
    [a : int
4     b : int
     c : int]
6   body
    [a = 33
     x = a-1 (* = 32 *)
     b = a-x (* = 1 *)
     c = x-b] (* = 31 *)
let a = 10
12 in ((from m1 take a) - (from m1 take b))-a
```

5.1.1 Concrete Syntax

A program in SIMPLE-MODULES consists of a possible empty sequence of module declarations followed by an expression:

```
\langle Program \rangle ::= \{\langle ModuleDefn \rangle\}^* \langle Expression \rangle
```

Expressions are the those of REC but with an extra production that we refer to as a $\frac{\text{qualified}}{\text{variable reference}}$

```
⟨Expression⟩ ::= from ⟨Identifier⟩ take ⟨Identifier⟩
```

The concrete syntax of modules is given by the following grammar:

```
\begin{split} &\langle \mathsf{ModuleDefn} \rangle & ::= & \mathsf{module} \ \langle \mathsf{Identifier} \rangle \ \mathsf{interface} \ \langle \mathsf{Iface} \rangle \ \mathsf{body} \ \langle \mathsf{ModuleBody} \rangle \\ &\langle \mathsf{Iface} \rangle & ::= & [\{\langle \mathsf{Decl} \rangle\}^*] \\ &\langle \mathsf{Decl} \rangle & ::= & \langle \mathsf{Identifier} \rangle : \langle \mathsf{Type} \rangle \\ &\langle \mathsf{ModuleBody} \rangle & ::= & [\{\langle \mathsf{Defn} \rangle\}^*] \\ &\langle \mathsf{Defn} \rangle & ::= & \langle \mathsf{Identifier} \rangle = \langle \mathsf{Expression} \rangle \end{split}
```

5.1.2 Abstract Syntax

```
type expr =
      | QualVar of string*string
   and
     texpr =
      IntType
        BoolType
       UnitType
      FuncType of texpr*texpr
        RefType of texpr
   and
12
     vdecl = string*texpr
   and
     vdef = string*expr
   type interface = ASimpleInterface of vdecl list
   type module_body = AModBody of vdef list
type module_decl = AModDecl of string*interface*module_body
   type prog = AProg of (module_decl list)*expr
```

5.2 Interpreter

5.2.1 Specification

The set of results is the same as that for EXPLICIT-REFS. In particular, the set of expressed values consists of integers, booleans, unit, closures and locations:

$$\mathbb{E} \mathbb{V} := \mathbb{Z} \cup \mathbb{B} \cup \mathbb{U} \cup \mathbb{CL} \cup \mathbb{L}$$

There are three kinds of evaluation judgements in SIMPLE-MODULES, one for programs, one for expressions, and a third auxiliary one used to evaluate module definitions. The evaluation judgement for programs is:

AProg(mdecls,e),
$$\rho$$
, $\sigma \Downarrow r$, σ'

A program AProg(mdecls,e) consists of a sequence of module declarations mdecl and a main expression e. Also, ρ is the initial environment and σ the initial store. The result of the evaluation is r and the updated store is σ' . Evaluation judgements for expressions are similar to those of EXPLICIT-REFS:

$$e, \rho, \sigma \downarrow r, \sigma'$$

$$\frac{\mathsf{mdecls}, \rho, \sigma \Downarrow \rho', \sigma' \quad \mathsf{e}, \rho', \sigma' \Downarrow r, \sigma''}{\mathsf{AProg}(\mathsf{mdecls}, \mathsf{e}), \rho, \sigma \Downarrow r, \sigma''} \, \mathsf{EProg}$$

$$\frac{-}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma} \, \mathsf{EMDeclsEmpty}$$

$$\frac{\mathsf{body}, \rho, \sigma \Downarrow \rho', \sigma' \quad \mathsf{ms}, \rho \oplus \{\mathsf{id} := \rho'\}, \sigma' \Downarrow \rho'', \sigma''}{\mathsf{AModDecl}(\mathsf{id}, \mathsf{iface}, \mathsf{body}) \quad \mathsf{ms}, \rho, \sigma \Downarrow \rho'', \sigma''} \, \mathsf{EMDeclsCons}$$

$$\frac{-}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma} \, \mathsf{EBValsEmpty}$$

$$\frac{\mathsf{e}, \rho, \sigma \Downarrow v, \sigma' \quad \mathsf{vs}, \rho \oplus \{\mathsf{id} := v\}, \sigma' \Downarrow \rho', \sigma''}{(\mathsf{id}, \mathsf{e}) \quad \mathsf{vs}, \rho, \sigma \Downarrow \rho', \sigma''} \, \mathsf{EBValsCons}$$

$$\frac{-}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma \Downarrow \rho', \sigma''} \, \mathsf{EBValsCons}$$

$$\frac{-}{\epsilon, \rho, \sigma \Downarrow \rho, \sigma \Downarrow \rho', \sigma''} \, \mathsf{EQualVar}(\mathsf{mid}, \mathsf{vid}), \rho, \sigma \Downarrow v, \sigma} \, \mathsf{EQualVar}(\mathsf{mid}, \mathsf{vid}), \rho, \sigma \Downarrow v, \sigma}$$

Figure 5.1: Evaluation Semantics for SIMPLE-MODULES (error propagation rules omitted)

where e is an expression, ρ an environment, σ the initial store, r the result and σ' the final store. The difference is that the environment will also allow for mappings between module identifiers and their bodies. The body of a module will be implemented as an environment too. The third evaluation judgement is:

$$\mathsf{mdecls}, \rho, \sigma \Downarrow \rho', \sigma'$$

Here mdecls is a sequence of module declarations, ρ is an initial environment and σ is an initial store. Evaluation of mdecls will produce an environment ρ' which associates to each module mid in mdecls an environment ρ_{mid} . Evaluation also produces an updated store σ' .

5.2.2 Implementation

We first extend environments to support bindings for modules:

Evaluation of programs consists in first evaluating all module definitions producing an environment as a result, and then evaluating the main expression using this environment. The former is achieved with the helper function eval_module_definitions : module_decl list -> env ea_result. We'll describe this function shortly.

```
let eval_prog (AProg(ms,e)) : exp_val ea_result =
  eval_module_definitions ms >>+
  eval_expr e
```

Evaluation of expressions is just like in EXPLICIT-REFS, except that we must deal with the new case, namely that of a qualified variable QualVar(module_id,var_id).

```
let rec eval_expr : expr -> exp_val ea_result =
   fun e ->
match e with
...
l QualVar(module_id,var_id) ->
   apply_env_qual module_id var_id
...
```

The helper function apply_env_qual inspects the environment looking for a module name module_id and then an identifier var_id declared within that module:

```
let rec apply_env_qual : string -> string -> exp_val ea_result =
    fun mid id ->
fun env ->
match env with

EmptyEnv -> Error "Key not found"
ExtendEnv(key,value,env) -> apply_env_qual mid id env
ExtendEnvRec(key,param,body,env) -> apply_env_qual mid id env
ExtendEnvMod(moduleName,bindings,env) ->
if mid=moduleName
then apply_env id bindings
else apply_env_qual mid id env
```

Finally, we turn to the above mentioned eval_module_definitions helper function. Given a list of module declarations ms, it evaluates them one by one using eval_module_definition and then returning a value of type env ea_result holding the resulting environment.

```
let rec eval_expr : expr -> exp_val ea_result =
   and
3
     eval_module_definition : module_body -> env ea_result =
     fun (AModBody vdefs) ->
     lookup_env >>= fun glo_env -> (* holds all previously declared modules *)
     List.fold_left
        (fun loc_env (var,decl) ->
            loc_env >>+
            (append_env_rev glo_env >>+
             eval_expr decl >>=
11
             extend_env var))
         (empty_env ())
13
15
     eval_module_definitions : module_decl list -> env ea_result =
17
     fun ms ->
     List.fold_left
       (fun curr_en (AModDecl(mname, minterface, mbody)) ->
          curr_en >>+
          (eval_module_definition mbody >>=
21
           extend_env_mod mname))
       lookup_env
```

```
25
    and
      eval_prog (AProg(ms,e)) : exp_val ea_result =
      eval_module_definitions ms >>+
27
      eval_expr e
                                                                              interp.ml
   utop # interp "
   module m1
     interface
       [u : int]
     body
      [u = 44]
6
   module m2
8
     interface
       [v : int]
10
     body
      [v = (from m1 take u)-11]
   let a=zero?(0)
      in debug(0)";;
   Environment:
14
   (a, BoolVal true)
   Module m2[(v,NumVal 33)]
   Module m1[(u,NumVal 44)]
   Store:
   Empty
   - : Ds.exp_val ReM.result = ReM.Ok Ds.UnitVal
                                                                                   utop
```

5.3 Type-Checking

5.3.1 Specification

```
\begin{array}{lll} \mbox{Judgements for typing programs} & \mbox{$\vdash$ AProg(ms,e):$} t \\ \mbox{Judgements for typing expressions} & \Delta; \Gamma \mbox{$\vdash$ e:$} t \\ \mbox{Judgements for typing list of module declarations} & \Delta_1 \mbox{$\vdash$ ms:$} \Delta_2 \end{array}
```

 Γ is the standard type environment from before Δ is a module type environment and is required for typing the expression from m take x

A module type is an expression of the form $m[u_1:t_1,\ldots,u_n:t_n]$. A module type environment is a sequence of module types.

$$\Delta ::= \epsilon \mid \mathbf{m}[u_1:t_1,\ldots,u_n:t_n] \Delta$$

We use letters Δ to denote module type environments. The empty module type environment is written ϵ . If $\mathtt{m}[u_1:t_1,\ldots,u_n:t_n]\in\Delta$, then we $\mathtt{m}\in\mathsf{dom}(\Delta)$. Moreover, in that case, have $u_i\in\mathsf{dom}(\Delta(\mathtt{m}))$, for $i\in 1..n$, and also $\Delta(\mathtt{m},u_i)=t_i$.

There is just one typing rule for typing programs, namely TProg. There is one new typing rule for expressions, namely TFromTake, it allows to type qualified variables. There are two typing rules for lists of module definitions: one for when the list is empty (TModE) and one for when it is not (TModNE). Regarding the latter,

```
\frac{\epsilon \vdash \mathtt{ms} :: \Delta \quad \Delta; \epsilon \vdash \mathtt{e} :: t}{\vdash \mathtt{AProg}(\mathtt{ms}, \mathtt{e}) :: t} \mathsf{TProg} \frac{\mathtt{m} \in \mathsf{dom}(\Delta) \quad x \in \mathsf{dom}(\Delta(\mathtt{m})) \quad \Delta(\mathtt{m}, \mathtt{x}) = \mathtt{t}}{\Delta; \Gamma \vdash \mathsf{from} \ \mathtt{m} \ \mathsf{take} \ \mathtt{x} :: t} \mathsf{TFromTake} \frac{\Delta; \Gamma \vdash \mathsf{from} \ \mathtt{m} \ \mathsf{take} \ \mathtt{x} :: t}{\Delta \vdash \epsilon :: \epsilon} \mathsf{TModE} (\Delta_1; [y_1 := s_1] \dots [y_{j-1} := s_{j-1}] \Gamma \vdash e_j :: s_j)_{j \in J} (t_i = s_{f(i)})_{i \in I} (t_i = s_{f(i)})_{i \in I} \mathsf{m}[x_i : t_i]_{i \in I} \Delta_1 \vdash \mathsf{ms} :: \Delta_2 \Delta_1 \vdash \mathtt{m}[x_i : t_i]_{i \in I} [y_j = e_j]_{j \in J} \mathsf{ms} :: \mathtt{m}[x_i : t_i]_{i \in I} \Delta_2 \mathsf{TModNE}
```

Figure 5.2: Typing rules for SIMPLE-MODULES

- Δ_2 is the type of the list of modules ms
- $m[x_i:t_i]_{i\in I}\Delta_1$ is the type of the list of modules that ms can use
- $[x_i]_{i\in I} \lhd [y_j]_{j\in J}$ means that the list of variables $[x_i]_{i\in I}$ is a sublist of the list of variables $[y_j]_{j\in J}$. This relation determines an injective, order preserving function $f:I\to J$

5.3.2 Implementation

```
type tenv =
     | EmptyTEnv
     ExtendTEnv of string*texpr*tenv
     ExtendTEnvMod of string*tenv*tenv
                                                                              dst.ml
   let rec
    type_of_prog (AProg (ms,e)) =
     type_of_modules ms >>+
     chk_expr e
     type_of_modules : module_decl list -> tenv tea_result =
     fun mdecls ->
     List.fold_left
       (fun curr_tenv (AModDecl(mname, ASimpleInterface(expected_iface), mbody)) ->
         curr_tenv >>+
10
         (type_of_module_body mbody >>= fun i_body ->
         if (is_subtype i_body expected_iface)
12
           extend_tenv_mod mname (var_decls_to_tenv expected_iface)
14
           error("Subtype failure: "^mname))
```

```
lookup_tenv
       {\tt mdecls}
20
     type_of_module_body : module_body -> tenv tea_result =
     fun (AModBody vdefs) ->
22
     lookup_tenv >>= fun glo_tenv ->
     (List.fold_left (fun loc_tenv (var,decl) ->
24
          loc_tenv >>+
          (append_tenv_rev glo_tenv >>+
26
          chk_expr decl >>=
          extend_tenv var))
28
         (empty_tenv ())
         vdefs) >>= fun tmbody ->
30
     return (reverse_tenv tmbody)
32
   and
     chk_expr : expr -> texpr tea_result =
     fun e ->
34
     match e with
     | Int n -> return IntType
36
     | Var id -> apply_tenv id
     | QualVar(module_id,var_id) ->
      apply_tenv_qual module_id var_id
```

5.4 Further Reading

Module inclusion Private types First-class modules

Appendix A

Supporting Files

A.1 File Structure

Typical file structure for an interpreter (in this example, ARITH).

The source files are in the src directory and the unit tests are in the test directory.

```
ast.ml Abstract Syntax

ds.ml Supporting data structures including expressed values, environments and results interp.ml Interpreter

lexer.mll Lexer generator

parser.mly Parser generator

test.ml Unit tests
.ocamlinit Loaded by utop upon execution; opens some modules
```

We use the dune build system for OCaml. You can find documentation on dune at https://readthedocs.org/projects/dune/downloads/pdf/latest/. Some common dune commands:

	D :1 1	. 1		- 1	. 1		
•	Build	the	project	and	then	run	utop

\$ dune utop

• Build the project

\$ dune build

• Clean the current project (erasing _build directory)

\$ dune clean

• Run tests (building if necessary)

\$ dune runtest

• Generate documentation (install odoc with opam first):

\$ dune build @doc

The generated html files are in:

\$ open _build/default/_doc/_html/index.html

Appendix B

Solution to Selected Exercises

Section 2.1

Answer B.0.1 (Exercise 2.2.1). Sample expressions of each of the following types are:

```
    expr. An example is: Int 2.
    env. Examples are: EmptyEnv and ExtendEnv("x", NumVal 2, EmptyEnv)
    exp_val. Examples are: NumVal 3 and BoolVal true.
```

- 4. exp_val result. Examples are: Ok (NumVal 3) and Ok (BoolVal true) and Error "oops".
- 5. int result. Examples are: Ok 1 and Error "oops".
- 6. env result. Examples are: Ok EmptyEnv and Ok (ExtendEnv("x", NumVal 2, EmptyEnv)).
- 7. int ea_result. Examples are: return 2 and error "oops".
- 8. exp_val ea_result. Examples are: return (NumVal 2) and return (BoolVal true) and error "oops".

 Also, apply_env "x".
- 9. env ea_result. Examples are: return (EmptyEnv) and error "oops". Also, extend_env "x" (NumVal 7).

Section 2.3

```
Answer B.0.2 (Exercise 2.4.8).
```

```
let even = proc (e) { proc (o) { proc (x) {
        if zero?(x)
        then zero?(0)
4      else (((o e) o) (x-1)) }};
in
6  let odd = proc(e) { proc (o) { proc (x) {
        if zero?(x)
8      then zero?(1)
        else (((e e) o) (x-1)) }};
in (((even even) odd) 4)
```

```
Answer B.0.3 (Exercise 2.4.9).

let f =

proc (g) {

proc (leaf) {

proc (depth) {

if zero?(depth) then (leaf, leaf)

else ((((g g) leaf) (depth-1)), (((g g) leaf) (depth-1))) }};

in let pbt = proc (leaf) { proc (height) { (((f f) leaf) height) }};

in ((pbt 2) 3)
```

Section 2.5

```
Answer B.0.4 (Exercise 2.5.2).
```

```
# interp
   let l1 = cons(1, cons(2, cons(3, emptylist)))
   in let l2 = cons(4, cons(5, emptylist))
  in letrec append(l1) = proc (l2) {
                             if empty?(l1)
                             then 12
6
                             else cons(hd(l1),((append\ tl(l1))\ l2))
   in ((append l1) l2)
   - : exp_val Rec.Ds.result =
   Ok (ListVal [NumVal 1; NumVal 2; NumVal 3; NumVal 4; NumVal 5])
   # interp "
   let l = cons(1, cons(2, cons(3, emptylist)))
   in let succ = proc(x) \{x+1\}
   in letrec map(l) = proc (f) {
                         if empty?(l)
                         then\ emptylist
20
                         else cons((f hd(l)), ((map tl(l)) f))
22
   in ((map l) succ)
   ";;
   - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 2; NumVal 3; NumVal
   4])
   # interp "
   let l = cons(1, cons(2, cons(1, emptylist)))
   in let is_one = proc (x) { zero?(x-1) }
   in letrec filter(l) = proc (p) {
                            if empty?(l)
34
                            then emptylist
                            else (if (p \ hd(l))
                                  then cons(hd(l),((filter\ tl(l))\ p))
                                  else ((filter tl(l)) p))
   in ((filter l) is_one)";;
   - : exp_val Rec.Ds.result = Ok (ListVal [NumVal 1; NumVal 1])
                                                                                  utop
```

Answer B.0.5 (Exercise 2.5.3). 1. debug Must be placed in the then case:

```
let z = 0
in let prod = proc (x) { proc (y) { x*y }}
in letrec f(n) = if zero?(n) then debug(1) else ((prod n) (f (n-1)))
in (f 10)
```

2. Two possible solutions are:

```
let z = 0

2 in debug(let prod = proc (x) { proc (y) { x*y }}

in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))

in (f 10))
```

or:

```
let z = 0
in let prod = debug(proc (x) { proc (y) { x*y }})
in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))
in (f 10)
```

3. debug must be placed in the body of proc (x):

```
let z = 0

in let prod = proc (x) { debug(proc (y) { x*y })}

in letrec f(n) = if zero?(n) then 1 else ((prod n) (f (n-1)))

in (f 10)
```

Bibliography

[Fri08] Daniel P. Friedman. <u>Essentials of Programming Languages</u>. The MIT Press, 3rd edition, 2008